
この資料は英語版を翻訳したもので、内容に相違が
ください。設計の際には、最新の英語版で内容をご

QII51007-6.0.0
生じる場合には原文を優先します。こちらの日本語版は参考用としてご利用

6. 推奨される HDL
コーディング構文
確認ください。
はじめに HDLコーディング・スタイルは、プログラマブル・ロジック・デザイン
で得られる結果の品質に大きく影響する場合があります。合成ツールは、
ロジック利用率とパフォーマンスの両面でHDLコードを最適化します。
ただし、場合によっては、最適化を最大限に実行するために人間による
デザインの理解が要求されたり、合成ツールにデザインの目的や意図に
ついての情報がまったく与えられないこともあります。最適化において
結果の品質を改善できる機会は、数多くあります。

この章では、アルテラのデバイスをターゲットとしたときに、最適な合
成結果を得るための HDL コーディング・スタイルの推奨事項について
説明します。以下の項から構成されます。

■ アルテラのメガファンクション
■ HDLコードでのアルテラ・メガファンクションのインスタンス化
■ HDLコードからのアルテラ・メガファンクションの推測
■ デバイス固有のコーディング・ガイドライン
■ その他のロジック構造のコーディング・ガイドライン

デザインの構造化のガイドラインについて詳しくは、「Quartus IIハンド
ブック Volume 1」の「Design Recommendations for Altera Devices」の
章を参照してください。

スタイルの推奨事項、オプション、または合成ツール（Quartus® II合成
機能とその他のサードパーティEDAツールを含む）に固有の HDL属性
については、ツール・ベンダのマニュアル、または「Quartus II ハンド
ブック Volume 1」の「合成」セクションの該当する章を参照してくださ
い。

Quartus IIハンドブック Volume 1
アルテラの
メガファンク
ションの使用

アルテラでは、アルテラ・デバイスのアーキテクチャ用に最適化された、
パラメータ化されたメガファンクションを提供しています。独自のロ
ジック・コーディングの代わりにメガファンクションを使用することで、
貴重なデザイン時間が節約できます。さらに、アルテラが提供するメガ
ファンクションを使用すると、より効率的にロジック合成とデバイス実
装を行うことができます。メガファンクションのサイズをスケーリング
したり、またパラメータを設定することによって各種オプションを設定
できます。メガファンクションには、パラメータ化されたモジュールの
ライブラリ（LPM）とアルテラ・デバイス固有のメガファンクションが
あります。

メモリ、DSP ブロック、LVDS デバイス、PLL（phase-locked
loops）、トランシーバ、ダブル・データ・レート入力 / 出力
（DDIO）回路などの、アルテラ・デバイス固有の一部の機能にア
クセスする場合、メガファンクションを使用する必要があります。

HDLコードでメガファンクションを使用するには、6–3ページの「HDL
コードでのアルテラ・メガファンクションの インスタンス化」で説明す
るようにインスタンス化します。デバイス・ファミリまたはベンダに依
存しない独自のコードを作成する方が望ましい場合や、メガファンク
ションを直接インスタンス化したくない場合があります。メガファンク
ションのインスタンス化を希望しない場合は、6–6ページの「HDLコー
ド からの アルテラ・ メガファンクションの推測」のガイドラインおよび
コーディング例に従い、汎用 HDL コードで適切なアルテラ・メガファ
ンクションが推測されるようにします。 

一部のデザインでは、メガファンクションをインスタンス化するより、
汎用HDLコードを使用した方が良い結果が得られることがあります。標
準 HDL コードを使用する場合とメガファンクションを使用する場合を
説明した、以下の一般的なガイドラインと例を参照してください。

■ 簡単な加算または減算ファンクションについては、LPM ファンクショ
ンではなく + または - シンボルを使用してください。簡単な算術演
算に対して LPMファンクションをインスタンス化した場合は、ファ
ンクションがハードコードされ、合成アルゴリズムが基本的なロジッ
ク最適化を活用できなくなるため、効率の高い結果が得られないこ
とがあります。同期ローダブル・カウンタなどの複雑な算術ロジッ
クの場合は、LPMファンクションを使用すると、HDLコードからは
推測が困難なアーキテクチャ固有の詳細な機能にアクセスできます。

■ 簡単なマルチプレクサおよびデコーダの場合、LPM ファンクション
の代わりにアレイ表記法（out = data[sel]など）を使用してく
ださい。アレイ表記法は非常に有効に機能し構文も簡単です。APEX™

シリーズ・デバイスのカスケード・チェインなどのアーキテクチャ
機能を活用するには LPM_MUXファンクションを使用できますが、こ

HDLコードでのアルテラ・メガファンクションの インスタンス化
の LPMファンクションは、特定の実装を強制する場合にのみ使用し
てください。

■ 可能であれば、除算演算は避けてください。除算は本質的に低速な
演算です。設計者の多くは乗算を工夫して、除算結果を得ています。
除算が必要な場合は、LPM_DIVIDE ファンクションを使用すると可
能な最良の結果が得られます。

HDLコードで
のアルテラ・
メガファンク
ションの
インスタンス
化

以下の項では、HDLコードのメガファンクションを以下の方法でインス
タンス化して使用する方法を説明します。

■ MegaWizard® Plug-In Manager を使用したメガファンクションのイ
ンスタンス化—MegaWizard Plug-In Managerを使用すると、ファン
クションをパラメータ化し、ラッパ・ファイルを作成できます。

■ サードパーティ合成ツール用クリア・ボックス・ネットリスト・ファ
イルの作成 — オプションで、ラッパ・ファイルの代わりにクリア・
ボックス・ボディを作成することができます。

■ ポート & パラメータ定義を使用したメガファンクションのインスタン
ス化—HDLコードで直接ファンクションをインスタンス化できます。

MegaWizard Plug-In Managerを使用したメガファンクショ
ンのインスタンス化

HDL コードでインスタンス化できるメガファンクションを Quartus II
GUI で作成するには、この項の説明に従って MegaWizard Plug-In
Managerを使用します。MegaWizard Plug-In Managerは、メガファン
クションをカスタマイズおよびパラメータ化するためのグラフィカル・
ユーザ・インタフェースを提供し、すべてのメガファンクション・パラ
メータを正しく設定できるようにします。パラメータの設定が終了した
ら、生成するファイルを指定できます。選択した言語に応じて、
MegaWizard Plug-In Manager は正しいパラメータでメガファンクショ
ンをインスタンス化し、以下のファイル・セットのいずれかを生成しま
す。

■ AHDL テキスト・デザイン・ファイル（.tdf）ラッパ・ファイルおよ
びサンプルのインスタンス・テンプレートText Design File（_inst.tdf）。

■ Verilog HDL（.v）ラッパ・ファイル、サンプルのインスタンス・テ
ンプレート Verilog HDLファイル（_inst.v）、およびブラック・ボッ
クス Verilog HDLモジュール宣言。

■ VHDL（.vhd）ラッパ・ファイルおよびサンプルのインスタンス・テ
ンプレート VHDLファイル（_inst.vhd）。

デザイン内のメガファンクション・ラッパ・ファイルは、対応するサン
プルのインスタンス・ファイルを使用してインスタンス化できます。ま
た、MegaWizard Plug-In Managerはデフォルトで、以下のデフォルト・
ファイルも作成します。

■ VHDL デザイン・ファイルで使用できるコンポーネント宣言ファイ
ル（.cmp）

■ テキスト・デザイン・ファイル（.tdf）で、または Verilog HDLデザ
イン・ファイルの参照用として使用できるADHLインクルード・ファ
イル（.inc）

MegaWizard Plug-In Manager で生成されるファイルのリストと説明に
ついては、表 6–1を参照してください。

表 6–1.  MegaWizard Plug-In Manager生成のファイル（ １ /２）

ファイル 説明

<出力ファイル >.bsf ブロック・シンボル・ファイル —Quartus II ブロック・デザイン・ファイル
（.bdf）で使用されます。

<出力ファイル >.cmp コンポーネント宣言ファイル—VHDLデザインで使用されます。

<出力ファイル >.inc ADHLインクルード・ファイル—AHDLデザインで使用されます。

<出力ファイル >.tdf(1) AHDLラッパ・ファイル—AHDLデザインでインスタンス化するためのメガファ
ンクション・ラッパ・ファイル

<出力ファイル >.vhd(2)(4) VHDLラッパ・ファイル—VHDLデザインのインスタンス化のためのメガファ
ンクション・ラッパ・ファイル、またはクリア・ボックス・ネットリスト・
ファイル。

<出力ファイル >.v(3)(4) Verilog HDLラッパ・ファイル—Verilog HDLデザインのインスタンス化のため
のメガファンクション・ラッパ・ファイル、またはクリア・ボックス・
ネットリスト・ファイル。

<出力ファイル >_bb.v(3) ブラック・ボックス Verilog HDLモジュール宣言—サードパーティ合成ツール
でブラック・ボックスを作成する場合、ポートの方向を指定するために Verilog
HDLデザインで使用される中空モジュール宣言。

<出力ファイル >_inst.tdf(1) テキスト・デザイン・ファイル・インスタンス・テンプレート—メガファンク
ション・ラッパ・ファイルにあるサブデザインのサンプル AHDLインスタンス。

<出力ファイル>_inst.vhd(2) VHDLインスタンス・テンプレート—メガファンクション・ラッパ・ファイル
にあるエンティティのサンプル VHDLインスタンス。

サードパーティ合成ツール用クリア・ボックス・ネットリスト・
ファイルの作成

サードパーティ合成ツールで特定のメガファンクションを使用する場
合、オプションでラッパ・ファイルの代わりにクリア・ボックス・ボディ
を作成できます。クリア・ボックス・ボディは完全に合成されたメガファ
ンクションで、特定のサードパーティ EDA合成ツールで使用できます。
メガファンクション・クリア・ボックス・ボディを含むネットリスト・
ファイルは、サードパーティ合成ツール向けに、Quartus II ソフトウェ
アで使用されるアーキテクチャ上の詳細情報を提供します。特定の合成
ツールはこの情報を利用することで、タイミングおよびリソース利用率
の予測をより正しく報告できます。また、合成ツールはタイミング情報
を使用して、タイミング・ドリブン最適化に集中できます、また結果の
品質を向上させることができます。

合成ツールでのクリア・ボックスのサポートについて詳しくは、ツール・
ベンダのマニュアル、または「Quartus II ハンドブック Volume 1」の
「合成」セクションの該当する章を参照してください。

クリア・ボックス・ネットリストを生成するには、MegaWizard Plug-In
Manager のメガファンクション選択ページ 2a で、Generate clear box
netlist file instead of a default wrapper file (for use with supported
EDA synthesis tools only)をオンにします。

<出力ファイル >_inst.v(3) Verilog HDLインスタンス・テンプレート—メガファンクション・ラッパ・
ファイルにあるモジュールのサンプル Verilog HDLインスタンス。

表 6–1の注 :
(1) MegaWizard Plug-In Managerは、AHDL出力ファイルを選択した場合にのみこのファイルを生成します。 
(2) MegaWizard Plug-In Managerは、VHDL出力ファイルを選択した場合にのみこのファイルを生成します。 
(3) MegaWizard Plug-In Manager は、Verilog HDL 出力ファイルを選択した場合にのみこのファイルを生成
します。 

(4) メガファンクション・ラッパ・ファイルは、ほとんどのメガファンクションに対してデフォルトで作成さ
れます。クリア・ボックスの機能を活用するには、ToolsメニューでMegaWizard Plug-In Managerをク
リックし、Generate clear box netlist file instead of a default wrapper file (for use with supported EDA
synthesis tools only) をオンにします。MegaWizard Plug-In Manager の使用方法について詳しくは、
Quartus IIヘルプを参照してください。

表 6–1.  MegaWizard Plug-In Manager生成のファイル（ ２ /２）

ファイル 説明

すべてのメガファンクションがクリア・ボックス・ネットリスト
をサポートしているわけではありません。特定のメガファンク
ションに対してクリア・ボックス・ネットリストを作成できない
場合、MegaWizard Plug-In Managerのページ 2aにはネットリス
トの生成オプションが表示されません。常にクリア・ボックス・
ネットリスト・ファイルを使用するメガファンクションもあり、
その場合はページ2aのオプションをオフにすることはできません。

ポートおよびパラメータ定義を使用したメガファンクショ
ンのインスタンス化

メガファンクションは、他のサブデザイン、モジュール、またはコンポー
ネントのように、メガファンクションを呼び出してパラメータを設定す
ることによって、AHDL、Verilog HDL、または VHDLコードで直接イ
ンスタンス化できます。

メガファンクションのポートとパラメータのリストについては、Quartus II
ヘルプの該当するメガファンクションを参照してください。Quartus II
ヘルプには、VHDLコンポーネント宣言のサンプルと、各メガファンク
ションの AHDLファンクションのプロトタイプも記載されています。

アルテラでは、PLL、トランシーバ、LVDS ドライバなどの複雑
なメガファンクションには、MegaWizard Plug-In Managerの使
用を推奨しています。MegaWizard Plug-In Managerの使用方法
について詳しくは、6–3ページの「MegaWizard Plug-In Manager
を使用したメガファンクションのインスタンス化」を参照してく
ださい。

HDLコード
からの
アルテラ・
メガファンク
ションの推測

Quartus II 合成機能を含む合成ツールは、特定のタイプの HDL コード
を認識し、適切なメガファンクションを自動的に推測します。合成ツー
ルは、デザインをコンパイルするときには、特にメガファンクションを
インスタンス化しない場合でも、アルテラ・メガファンクション・コー
ドを使用します。合成ツールは、アルテラ・デバイスに最適化されたロ
ジックを利用するメガファンクションを推測します。このようなロジッ
クの面積と性能は、同じ HDL コードの汎用ロジックを推測して得られ
る結果よりも良好な場合があります。

以下の項では、標準合成ツールが認識し、メガファンクションにマップ
するロジックのタイプについて説明します。合成ソフトウェアは、ここ
に挙げる特定のファンクションのみ推測します。合成ソフトウェアは、
PLL、LVDS ドライバ、トランシーバ、DDIO 回路など、その他のファ
ンクションをHDLコードから推測することはできません。これらのファ
ンクションは HDL コードで完全に、あるいは効率的に説明できないた
めです。合成ツール・オプションを使用して、特定のメガファンクショ

ンの推測をオフにできるケースもあります。以下の項では、以下のメガ
ファンクションを汎用 HDL コードから推測する方法について説明して
います。

■ lpm_mult—乗算器を HDLコードから推測
■ altmult_accum & altmult_add—乗算累積器および乗算加算器を

HDLコードから推測
■ altsyncram & lpm_ram_dp—RAMファンクションをHDLコードか
ら推測

■ lpm_rom—ROMを HDLコードから推測
■ altshift_taps—シフト・レジスタを HDLコードから推測

合成ツールの機能とオプションについては、合成ツールのマニュアル、
または「Quartus II ハンドブック Volume 1」の「合成」の項の該当する
章を参照してください。

lpm_mult—乗算器を HDLコードから推測

乗算器ファクションを推測する場合、合成ツールは乗算器を検索し、
lpm_multまたは altmult_addメガファンクションに変換します。あ
るいは乗算器デバイスの素子に直接マップすることもあります。DSPブ
ロックを含むデバイスの場合、ソフトウェアはデバイスの利用状況に応
じて、ロジックの代わりに乗算を DSP ブロックに実装できます。
Quartus IIフィッタは、入力および出力レジスタを DSPブロックに配置
（すなわち、レジスタ・パッキングを実行）し、性能および面積利用率を
改善することもできます。

DSP ブロックと DSP ブロックが実装できるファンクションについて詳
しくは、該当するアルテラ・デバイス・ファミリのハンドブックと、ア
ルテラWebサイト www.altera.comの DSPソリューション・センタを
参照してください。

以下の 4 つのコード・サンプルに、合成ツールが lpm_mult または
altmult_add メガファンクションとして推測できる符号なしおよび符
号付き乗算器の Verilog HDL と VHDL の例を示します。それぞれの例
は、1つの DSPブロックの 9ビット・エレメントにフィットします。ま
た、レジスタ・パッキングが実行されると、レジスタ用の特別なロジッ
ク・セルが不要になります。

Verilog HDLでの符号付き宣言は、Verilog 2001規格の 1つの機
能です。

例 6–1. Verilog HDL符号なし乗算器
module unsigned_mult (out, a, b);

output [15:0] out;
input  [7:0] a;
input  [7:0] b;
assign out = a * b;

endmodule

例 6–2.入力および出力レジスタを備えた Verilog HDL符号付き乗算器（パイプライン化 = 2）
module signed_mult (out, clk, a, b);

output [15:0] out;
input clk;
input signed [7:0] a;
input signed [7:0] b;

reg signed [7:0] a_reg;
reg signed [7:0] b_reg;
reg signed [15:0] out;
wire signed [15:0] mult_out;

assign mult_out = a_reg * b_reg;

always @ (posedge clk)
begin

a_reg <= a;
b_reg <= b;
out <= mult_out;

end
endmodule

HDLコード からの アルテラ・ メガファンクションの推測
例 6–3.入力および出力レジスタを備えた VHDL符号なし乗算器（パイプライン化 = 2）
LIBRARY ieee;
USE ieee.std_logic_1164.ALL;
USE ieee.numeric_std.all;

ENTITY unsigned_mult IS
PORT (

a: IN UNSIGNED (7 DOWNTO 0);
b: IN UNSIGNED (7 DOWNTO 0);
clk: IN STD_LOGIC;
aclr: IN STD_LOGIC;
result: OUT UNSIGNED (15 DOWNTO 0)

);
END unsigned_mult;

ARCHITECTURE rtl OF unsigned_mult IS
SIGNAL a_reg, b_reg: UNSIGNED (7 DOWNTO 0);

BEGIN
PROCESS (clk, aclr)
BEGIN

IF (aclr ='1') THEN
a_reg <= (OTHERS => '0');
b_reg <= (OTHERS => '0');
result <= (OTHERS => '0');

ELSIF (clk'event AND clk = '1') THEN
a_reg <= a;
b_reg <= b;
result <= a_reg * b_reg;

END IF;
END PROCESS;

END rtl;

例 6–4. VHDL符号付き乗算器
LIBRARY ieee;
USE ieee.std_logic_1164.ALL;
USE ieee.numeric_std.all;

ENTITY signed_mult IS
PORT (

a: IN SIGNED (7 DOWNTO 0);
b: IN SIGNED (7 DOWNTO 0);
result: OUT SIGNED (15 DOWNTO 0)

);
END signed_mult;

ARCHITECTURE rtl OF signed_mult IS
SIGNAL a_int, b_int: SIGNED (7 downto 0);
SIGNAL pdt_int: SIGNED (15 downto 0);

BEGIN
a_int <= (a);
b_int <= (b);
pdt_int <= a_int * b_int;
result <= pdt_int;

END rtl;

altmult_accum & altmult_add— 乗算累積器および乗算加
算器を HDLコードから推測

合成ツールは、乗算累積器または乗算加算器を検出し、それぞれ
altmult_accumまたは altmult_addメガファンクションに変換しま
す。この後 Quartus IIソフトウェアは、配置配線中にこれらのファンク
ションを DSPブロックに配置します。

合成ツールが乗算累積器と乗算加算器のファンクションを推測す
るのは、アルテラ・デバイス・ファミリが専用 DSP ブロックを
搭載している場合のみです。

乗算累積器は、加算器と乗算器から構成されます。加算器の出力は後段
のレジスタに供給され、そのレジスタ出力が加算器の入力となります。
乗算加算器は、1レベルまたは 2レベルの加算、減算、または加算 /減
算演算子に供給する 2～ 4個の乗算器から構成されます。加算が使用さ
れる場合には、常に第 2 レベルの演算子になります。Quartus II フィッ
タは、乗算累積器と乗算加算器に加えて、入力および出力レジスタを
DSPブロックに配置してレジスタをパックし、性能および面積利用率を
改善します。

例 6–5から 6–8に示す Verilog HDLと VHDLコードのサンプルでは、乗
算累積器と乗算加算器を推測しています。

例 6–5.入力、出力、およびパイプライン・レジスタを備えた Verilog HDL符号なし乗算累積器
（レイテンシ = 3）
module unsig_altmult_accum (dataout, dataa, datab, clk, aclr, clken);

input [7:0] dataa;
input [7:0] datab;
input clk;
input aclr;
input clken;
output [31:0] dataout;
reg [31:0] dataout;
reg [7:0] dataa_reg;
reg [7:0] datab_reg;
reg [15:0] multa_reg;
wire [15:0] multa;
wire [31:0] adder_out;
assign multa = dataa_reg * datab_reg;
assign adder_out = multa_reg + dataout;
always @ (posedge clk or posedge aclr)
begin

if (aclr)
begin

dataa_reg <= 0;
datab_reg <= 0;
multa_reg <= 0;
dataout <= 0;

end
else if (clken)
begin

dataa_reg <= dataa;
datab_reg <= datab;
multa_reg <= multa;
dataout <= adder_out;

end
end

endmodule

例 6–6. Verilog HDL符号付き乗算加算器（レイテンシ = 0）
module sig_altmult_add (dataa, datab, datac, datad, result);

input signed [15:0] dataa;
input signed [15:0] datab;
input signed [15:0] datac;
input signed [15:0] datad;
output [32:0] result;

wire signed [31:0] mult0_result;
wire signed [31:0] mult1_result;

assign mult0_result = dataa * datab;
assign mult1_result = datac * datad;
assign result = (mult0_result + mult1_result);

endmodule

例 6–7.入力、出力、およびパイプライン・レジスタを備えた VHDL符号なし乗算加算器
（レイテンシ = 3）
LIBRARY ieee;
USE ieee.std_logic_1164.ALL;
USE ieee.numeric_std.all;

ENTITY unsignedmult_add IS
PORT (

a: IN UNSIGNED (7 DOWNTO 0);
b: IN UNSIGNED (7 DOWNTO 0);
c: IN UNSIGNED (7 DOWNTO 0);
d: IN UNSIGNED (7 DOWNTO 0);
clk: IN STD_LOGIC;
aclr: IN STD_LOGIC;
result: OUT UNSIGNED (15 DOWNTO 0)

);
END unsignedmult_add;

ARCHITECTURE rtl OF unsignedmult_add IS
SIGNAL a_int, b_int, c_int, d_int: UNSIGNED (7 DOWNTO 0);
SIGNAL pdt_int, pdt2_int: UNSIGNED (15 DOWNTO 0);
SIGNAL result_int: UNSIGNED (15 DOWNTO 0);

BEGIN
PROCESS (clk, aclr)
BEGIN

IF (aclr = '1') THEN
a_int <= (OTHERS => '0');
b_int <= (OTHERS => '0');
c_int <= (OTHERS => '0');
d_int <= (OTHERS => '0');
pdt_int <= (OTHERS => '0');
pdt2_int <= (OTHERS => '0');
result_int <= (OTHERS => '0');

ELSIF (clk'event AND clk = '1') THEN
a_int <= a; 
b_int <= b; 
c_int <= c;
d_int <= d;
pdt_int <= a_int * b_int;
pdt2_int <= c_int * d_int;
result_int <= pdt_int + pdt2_int;

END IF;
END PROCESS;

result <= result_int;
END rtl;

例 6–8.入力、出力、およびパイプライン・レジスタを備えた VHDL符号付き乗算累積器
（レイテンシ = 3）
LIBRARY ieee;
USE ieee.std_logic_1164.ALL;
USE ieee.numeric_std.all;

ENTITY sig_altmult_accum IS
PORT (

a: IN SIGNED(7 DOWNTO 0);
b: IN SIGNED (7 DOWNTO 0);
clk: IN STD_LOGIC;
accum_out: OUT SIGNED (15 DOWNTO 0)

) ;
END sig_altmult_accum;

ARCHITECTURE rtl OF sig_altmult_accum IS
SIGNAL a_reg, b_reg: SIGNED (7 DOWNTO 0);
SIGNAL pdt_reg: SIGNED (15 DOWNTO 0);
SIGNAL adder_out: SIGNED (15 DOWNTO 0);

BEGIN
PROCESS (clk)
BEGIN

IF (clk'event and clk = '1') THEN
a_reg <= (a);
b_reg <= (b);

pdt_reg <= a_reg * b_reg;
adder_out <= adder_out + pdt_reg;

END IF;
  END process;

accum_out <= (adder_out);
END rtl;

altsyncram & lpm_ram_dp—RAMファンクションを HDL
コードから推測

合成ツールは専用の RAM ブロックを搭載したデバイス・ファミリに対
して、altsyncram または lpm_ram_dpメガファンクションに置換で
きるレジスタおよびロジックのセットを検出し、RAM ファンクション
を推測します。

合成ツールは、シングル・ポートおよびシンプル・デュアル・ポート（1
つのリード・ポートと 1つのライト・ポート）の RAMブロックを認識
します。汎用ロジックのレジスタを使用することによって、小さな RAM
ブロックを効率的に実装できるので、通常は小さな RAM ブロックは推
測しません。

Quartus II合成機能を使用している場合、すべてのサイズのRAM
ブロックを推測するよう指定できます。Assignmentsメニューの
Settings をクリックします。Category リストで、Analysis &
Synthesisをクリックします。More Settingsをクリックします。
Existing Options Settingsで、オプション Allow Any RAM Size
for Recognition を選択します。Setting 矢印をクリックし、ON
を選択します。

デザインに合成ツールが認識も推測もしない RAMブロックが含
まれている場合、デザインが大量のシステム・メモリを要求し、
これによってコンパイル問題が発生する可能性があります。

一部の合成ツールには、TriMatrixTM メモリ・ブロックを搭載するアル
テラ・デバイス用に、推測された RAM ブロックの実装を制御するオプ
ションが用意されています。例えば、Quartus II 合成機能では、値に
“M512”、“M4K”、または “M-RAM”を使用してメモリ・ブロックのタイ
プを指定するか、または値 “logic”を使用して、専用メモリ・ブロックの
代わりに通常のロジックの使用を指定するramstyle合成属性を使用しま
す。 

合成属性について詳しくは、「Quartus IIハンドブック Volume 1」の「合
成」セクションの該当する章を参照してください。

フォーマル検証フローを使用している場合、アルテラでは RAM ロジッ
クのみ搭載した別のエンティティまたはモジュールで RAM ブロックを
作成することを推奨しています。Quartus II 合成機能を使用する場合な
どの特定のフォーマル検証フローでは、フォーマル検証ツールは RAM
ブロックをサポートしないため、推測された RAM を持つエンティティ
またはモジュールは自動的にブラック・ボックスに置かれます。
Quartus II ソフトウェアは、このような場合は警告メッセージを発行し
ます。エンティティまたはモジュールの RAM ブロックの外に、追加ロ
ジックが搭載されている場合、このロジックもフォーマル検証ではブ
ラック・ボックスとして処理する必要があるため検証できません。

デュアル・クロック同期 RAM

アルテラの TriMatrix メモリ・ブロックは同期型です。このため、これ
らの専用メモリ・ブロックを含むアーキテクチャをターゲットとした
RAM デザインは、同期型でなければデバイス・アーキテクチャに直接
マップすることはできません。同期メモリは、すべてのアルテラ・デバ
イス・ファミリでサポートされています。

同じ RAM アドレスに対する読み出しと書き込みが同時に起こると、ア
ルテラ・デバイスの TriMatrix メモリ・ブロックは未定義のデータ値を
返します。この点はオリジナルの HDLデザインの機能とは異なります。
デザインで同じ RAM アドレスの読み出しと書き込みを行うときに特定
の出力が必要になる場合、合成ツールにこれらのメモリに対する RAM
推測を無効にすることによって、デュアル・クロック・メモリに対する
RAMブロックを推測しないように指示します。

合成ツールの RAM 推測を無効にする具体的なオプションについては、
合成ツールのマニュアルか、「Quartus IIハンドブック Volume 1」の「合
成」セクションの該当する章を参照してください。

例 6–9 および 6–10 のコード・サンプルには、デュアル・クロック同期
RAMを推測する Verilog HDLと VHDLコードを示しています。

例 6–9. Verilog HDLデュアル・クロック同期 RAM
module ram_dual (q, addr_in, addr_out, d, we, clk1, clk2);

output [7:0] q;
input [7:0] d;
input [6:0] addr_in;
input [6:0] addr_out;
input we, clk1, clk2;

reg [6:0] addr_out_reg;
reg [7:0] q;
reg [7:0] mem [127:0];

always @ (posedge clk1)
begin

if (we)
mem[addr_in] <= d;

end

always @ (posedge clk2) begin
q <= mem[addr_out_reg];
addr_out_reg <= addr_out;

end
endmodule

例 6–10. VHDLデュアル・クロック同期 RAM
LIBRARY ieee;
USE ieee.std_logic_1164.ALL;
ENTITY ram_dual IS

PORT (
clock1, clock2: IN STD_LOGIC;
data: IN STD_LOGIC_VECTOR (3 DOWNTO 0);
write_address: IN INTEGER RANGE 0 to 31;
read_address: IN INTEGER RANGE 0 to 31;
we: IN STD_LOGIC;
q: OUT STD_LOGIC_VECTOR (3 DOWNTO 0)

);
END ram_dual;
ARCHITECTURE rtl OF ram_dual IS

TYPE MEM IS ARRAY(0 TO 31) OF STD_LOGIC_VECTOR(3 DOWNTO 0);
SIGNAL ram_block: MEM;
SIGNAL read_address_reg : INTEGER RANGE 0 to 31;

BEGIN
PROCESS (clock1)
BEGIN

IF (clock1'event AND clock1 = '1') THEN
IF (we = '1') THEN

ram_block(write_address) <= data;
END IF;

END IF;
END PROCESS;
PROCESS (clock2)
BEGIN

IF (clock2'event AND clock2 = '1') THEN
q <= ram_block(read_address_reg);
read_address_reg <= read_address;

END IF;
END PROCESS;

END rtl;

Read-Through-Write動作のないシングル・クロック同期 RAM

この項のコード例には、シングル・クロック同期RAMを推測するVerilog
HDLと VHDLコードを示しています。アルテラの TriMatrixメモリ・ブ
ロックは同期型です。このため、これらの専用メモリ・ブロックを含む
アーキテクチャをターゲットとした RAM デザインは、同期型でなけれ
ばデバイス・アーキテクチャに直接マップすることはできません。

これらの例では、TriMatrixメモリ・ブロックで直接サポートされていな
い、read-through-write 動作も回避しています。アルテラでは、デザイ
ンで RAMに read-through-write動作が要求されない限り、すなわちデ
ザインで同じ RAM 位置に対する同時読み出しおよび書き込みで、現在
その RAM 位置に書き込み中の新しい値が要求されなければ、このコー
ディング・スタイルを使用することを推奨しています。

TriMatrix メモリ・ブロックでは、同一クロック・サイクルで同
じアドレスに対する読み出しと書き込みを試みた場合、メモリ・
モードとブロック・タイプに応じて、読み出し時にそのアドレス
の古いデータまたは未知のデータが返されます。

Read-through-write動作の RAMが必要な場合は、6–18ページの「Read-
Through-Write動作対応シングル・クロック同期 RAM」の項を参照して
ください。 

特定のデバイスの専用メモリ・ブロックについて詳しくは、アルテラ・
ウェブサイト www.altera.co.jpの該当するアルテラ・デバイス・ファミ
リのデータ・シートを参照してください。

例 6–11および 6–12に示すRAMコード・サンプルは、アルテラ TriMatrix
メモリに直接マップしています。

例 6–11. Read-Through-Write動作非対応 Verilog HDLシングル・クロック同期 RAM
module ram_infer (q, a, d, we, clk);
   output reg [7:0] q;
   input [7:0] d;
   input [6:0] a;
   input we, clk;

   reg [7:0] mem [127:0];

   always @ (posedge clk) begin
      if (we)
         mem[a] <= d;
      q <= mem[a]; // q doesn't get d in this clock cycle
   end
endmodule

例 6–12. Read-Through-Write非対応 VHDLシングル・クロック同期 RAM
LIBRARY ieee;
USE ieee.std_logic_1164.ALL;

ENTITY ram IS
   PORT (
      clock: IN STD_LOGIC;
      data: IN STD_LOGIC_VECTOR (2 DOWNTO 0);
      write_address: IN INTEGER RANGE 0 to 31;
      read_address: IN INTEGER RANGE 0 to 31;
      we: IN STD_LOGIC;
      q: OUT STD_LOGIC_VECTOR (2 DOWNTO 0)
   );
END ram;

ARCHITECTURE rtl OF ram IS
   TYPE MEM IS ARRAY(0 TO 31) OF STD_LOGIC_VECTOR(2 DOWNTO 0);
   SIGNAL ram_block: MEM;
BEGIN
   PROCESS (clock)
   BEGIN
      IF (clock'event AND clock = '1') THEN
         IF (we = '1') THEN
            ram_block(write_address) <= data;
         END IF;
         q <= ram_block(read_address); 
            -- VHDL semantics imply that q doesn't get data 
            -- in this clock cycle
      END IF;
   END PROCESS;
END rtl;

Read-Through-Write動作対応シングル・クロック同期 RAM

TriMatrixメモリ・ブロックは、混合ポートでの read-through-write動作
はサポートしていません。これは、同一クロック・サイクルで同じアド
レスに対する読み出しと書き込みを試みた場合、メモリ・モードとブロッ
ク・タイプに応じて、読み出し時にそのアドレスの古いデータまたは未
知のデータが返されることを意味します。ただし、同じ位置に対して同
時に読み出しと書き込みを行うと、読み出し時に現在その RAM 位置に
書き込み中の新しい値が読み出される HDLコードで、RAMブロックを
説明することができます。以下の例では、このタイプの RAM ロジック
を推測するコードを示しています。この働きをターゲット・デバイスに
実装するために、合成ソフトウェアは RAMブロックの周囲にバイパス・
ロジックを追加します。RAM ブロックがデザインのクリティカル・パ
スに含まれる場合、このバイパス・ロジックにより、デザインの面積利
用率が増加し性能が低下します。

例 6–13 および 6–14 に示す RAM 例では、RAM ブロックの周辺にバイ
パス・ロジックが必要です。

例 6–13. Read-Through-Write動作対応 Verilog HDLシングル・クロック同期 RAM
module ram_infer (q, a, d, we, clk);
    output [7:0] q;
    input [7:0] d;
    input [6:0] a;
    input we, clk;
reg [6:0] read_add;
    reg [7:0] mem [127:0];
always @ (posedge clk) begin
        if (we)
            mem[a] <= d;
        read_add <= a;
    end
assign q = mem[read_add];
endmodule

例 6–14. Read-Through-Write動作対応 VHDLシングル・クロック同期 RAM
LIBRARY ieee;
USE ieee.std_logic_1164.ALL;

ENTITY ram IS
PORT (

clock: IN STD_LOGIC;
data: IN STD_LOGIC_VECTOR (2 DOWNTO 0);
write_address: IN INTEGER RANGE 0 to 31;
read_address: IN INTEGER RANGE 0 to 31;
we: IN STD_LOGIC;
q: OUT STD_LOGIC_VECTOR (2 DOWNTO 0)

);
END ram;

ARCHITECTURE rtl OF ram IS
TYPE MEM IS ARRAY(0 TO 31) OF STD_LOGIC_VECTOR(2 DOWNTO 0);
SIGNAL ram_block: MEM;
SIGNAL read_address_reg: INTEGER RANGE 0 to 31;

BEGIN
PROCESS (clock)
BEGIN

IF (clock'event AND clock = '1') THEN
IF (we = '1') THEN

ram_block(write_address) <= data;
END IF;
read_address_reg <= read_address;

END IF;
END PROCESS;
q <= ram_block(read_address_reg);

END rtl;

2つのリード・アドレスを持つ同期 RAM

Quartus II 合成機能は、2 つのリード・アドレスと 1 つのライト・アド
レスを持つ RAM 記述から、RAM ブロックを推測できます。このタイ
プの RAM ブロックは、図 6-1 に示すように RAM ブロックを複製する
ことによって実装できます。ブロックごとに独立しているリード・アド
レスを除いて、両方の RAMブロックのすべての入力が複製されます。

図 6-1. 2つのリード・アドレスを持つ同期 RAMを示すブロック図

例 6–15と 6–16に示す 2つのリード・アドレスを持つ RAMコード・サ
ンプルは、RAMブロックの複製により推測されています。

例 6–15. 2つのリード・アドレスを持つ Verilog HDLシングル・クロック同期 RAM
module dual_ram_infer (q, q2, write_address, read_address, read_address2, d, we, clk);
      output reg [7:0] q;
      output reg [7:0] q2;
      input [7:0] d;
      input [6:0] write_address;
      input [6:0] read_address;
      input [6:0] read_address2;
      input we, clk;

      reg [7:0] mem [127:0];

      always @ (posedge clk) begin
            if (we)
                  mem[write_address] <= d;
            q <= mem[read_address];
            q2 <= mem[read_address2];
      end
endmodule

例 6–16. 2つのリード・アドレスを持つ VHDLシングル・クロック同期 RAM
LIBRARY ieee;
USE ieee.std_logic_1164.ALL;

ENTITY dual_ram_infer IS
  PORT (
    clock: IN STD_LOGIC;
    data: IN STD_LOGIC_VECTOR (2 DOWNTO 0);
    write_address: IN INTEGER RANGE 0 to 31;
    read_address: IN INTEGER RANGE 0 to 31;
    read_address2: IN INTEGER RANGE 0 to 31;
    we: IN STD_LOGIC;
    q: OUT STD_LOGIC_VECTOR (2 DOWNTO 0);
    q2: OUT STD_LOGIC_VECTOR (2 DOWNTO 0)
    );
END dual_ram_infer;

ARCHITECTURE rtl OF dual_ram_infer IS
  TYPE MEM IS ARRAY(0 TO 31) OF STD_LOGIC_VECTOR(2 DOWNTO 0);
  SIGNAL ram_block: MEM;
BEGIN
  PROCESS (clock)
  BEGIN
    IF (clock'event AND clock = '1') THEN
      IF (we = '1') THEN
        ram_block(write_address) <= data;
      END IF;
      q <= ram_block(read_address);
      q2 <= ram_block(read_address2);
    END IF;
  END PROCESS;
END rtl;
Altera Corporation 6–21
2006年 5月



Quartus IIハンドブック Volume 1
非同期リード・アドレスを持つシングル・クロック同期 RAM

例 6–17および 6–18に示すコード・サンプルに、非同期リード・アドレ
スおよびレジスタ出力を備えたRAMのVerilog HDLとVHDLコードを
示します。

以下の例の RAMコードの実装は、デバイス・ファミリの専用 RAMアー
キテクチャによって異なります。APEXシリーズのRAMアーキテクチャ
は非同期リード・アドレスをサポートしているため、例えば、APEX デ
バイス・シリーズに非同期リード・アドレスを実装するのは簡単です。
ただし、Stratix®デバイスおよび大部分の新しいデバイス・ファミリでは
リード・アドレスをラッチする必要があります。したがって、以下の例
では非同期 RAM コードを直接実装することはできません。Stratix シ
リーズのデバイスなどに非同期 RAM を実装する場合、合成ツールで出
力レジスタを RAM ブロックの入力に移動すれば、altsyncram メガ
ファンクションを使用してロジックを実装できます。リード・クロック
とライト・クロックが異なる場合、出力レジスタを RAM ブロックの入
力に移動すると機能が多少変更されることがあります。このような状況
では、合成ソフトウェアが警告を発行します。Quartus II 合成機能を使
用している場合、Quartus II ヘルプで違いを説明しています。これらの
RAMの例では、デバイスのアーキテクチャによっては直接 RAMブロッ
クにマップしない場合があります。

例 6–17.非同期リード・アドレスを持つ Verilog HDLシングル・クロック同期 RAM
module ram (clock, data, write_address, read_address, we, q);

parameter ADDRESS_WIDTH = 4;
parameter DATA_WIDTH    = 8;
input clock;
input [DATA_WIDTH-1:0] data;
input [ADDRESS_WIDTH-1:0] write_address;
input [ADDRESS_WIDTH-1:0] read_address;
input we;
output [DATA_WIDTH-1:0] q;

reg [DATA_WIDTH-1:0] q;
reg [DATA_WIDTH-1:0] ram_block [2**ADDRESS_WIDTH-1:0];
always @ (posedge clock)

begin
if (we)

ram_block[write_address] <= data;
q <= ram_block[read_address];

end
endmodule

例 6–18.非同期リード・アドレスを持つ VHDLシングル・クロック同期 RAM
LIBRARY ieee;
USE ieee.std_logic_1164.ALL;
USE ieee.numeric_std.ALL;

ENTITY ram IS
GENERIC (

ADDRESS_WIDTH: integer := 4;
DATA_WIDTH: integer := 8

);
PORT (

clock: IN  std_logic;
data: IN STD_LOGIC_VECTOR(DATA_WIDTH - 1 DOWNTO 0);
write_address IN  STD_LOGIC_VECTOR (ADDRESS_WIDTH - 1 DOWNTO 0);
read_address IN STD_LOGIC_VECTOR(ADDRESS_WIDTH - 1 DOWNTO 0);
we: IN STD_LOGIC;
q: OUT STD_LOGIC_VECTOR(DATA_WIDTH - 1 DOWNTO 0)

);
END ram;

ARCHITECTURE rtl OF ram IS
TYPE RAM IS ARRAY(0 TO 2 ** ADDRESS_WIDTH - 1) OF std_logic_vector(DATA_WIDTH - 1 

DOWNTO 0);
SIGNAL ram_block: RAM;

BEGIN
PROCESS (clock)
BEGIN

IF (clock'event AND clock = '1') THEN
IF (we = '1') THEN
    ram_block(TO_INTEGER(UNSIGNED(write_address))) <= data;
END IF;
q <= ram_block(TO_INTEGER(UNSIGNED(read_address)));

END IF;
END PROCESS;

END rtl;

初期メモリ内容の指定

合成ツールによっては、推測されるメモリの初期内容を指定できます。
例えば、Quartus II 合成機能は、推測される RAM ブロックにメモリ初
期化ファイル（.mif）を指定できる ram_init_fileという合成属性を
サポートしています。VHDLでは、対応する信号にデフォルト値を指定
することで、推測されるメモリの内容を初期化することもできます。
Quartus II合成機能では、推測される RAMのデフォルト値がMIFに自
動的に変換されます。

ram_init_file 属性について詳しくは、「Quartus II ハンドブック
Volume 1」の「Quartus II合成機能」の章を参照してください。他の合
成ツールの合成機能について詳しくは、ツール・ベンダのマニュアルを
参照してください。

lpm_rom—ROMを HDLコードから推測

ROMファンクションを推測する場合、合成ツールは専用の RAMブロッ
クを搭載したデバイス・ファミリに対してのみ、そのデバイス・ファミ
リに応じて altsyncramまたは lpm_romメガファンクションに置換で
きるレジスタおよびロジックのセットを検出します。

フォーマル検証ツールは ROMメガファンクションをサポートし
ないため、Quartus II 合成機能はサードパーティ・フォーマル検
証ツールが選択されている場合は、ROM メガファンクションを
推測しません。 

ROMが推測されるのは、case文内のすべての選択肢に定数値が設定さ
れている case文が存在する場合です。小規模な ROMは、一般的に通常
のロジックによるレジスタを使用して実装される場合に最高の性能を達
成するため、各 ROMファンクションが推測されメモリに配置されるに
は、最小サイズ要件を満たしている必要があります。

Quartus II 合成機能を使用している場合、ソフトウェアにすべて
のサイズの ROM ブロックを推測するよう指示できます。
Assignments メニューの Settings をクリックします。Category
リストで、Analysis & Synthesisをクリックします。More Settings
をクリックします。Existing Options Settings で、オプション
Allow Any ROM Size for Recognitionを選択します。Setting矢
印をクリックし、ONを選択します。

一部の合成ツールには、TriMatrix メモリ・ブロックを搭載するアルテ
ラ・デバイス用に、推測された ROMブロックの実装を制御するオプショ
ンが用意されています。例えば、Quartus II合成機能では、値に “M512”、
“M4K”、または “M-RAM”を使用してメモリ・ブロックのタイプを指定
するか、または値 “logic”を使用して、専用メモリ・ブロックの代わりに
通常のロジックの使用を指定する romstyle合成属性を使用します。

合成属性について詳しくは、「Quartus IIハンドブック Volume 1」の「合
成」セクションの該当する章を参照してください。フォーマル検証フロー
を使用している場合、アルテラでは ROMロジックのみ搭載した別のエ
ンティティまたはモジュールで ROMブロックを作成することを推奨し
ています。これはフォーマル検証中に、エンティティとモジュールをブ
ラック・ボックスとして処理する必要があるためです。

例 6–19および 6–20に示す Verilog HDLと VHDLコードのサンプルで
は、同期 ROMブロックを推測しています。デバイス・ファミリの専用
RAMアーキテクチャによっては、ROMロジックが同期型でなければな
りません。詳しくはデバイス・ファミリのハンドブックを参照してくだ
さい。 

Stratixシリーズ・デバイスや最近のデバイス・ファミリなど、同期 RAM
ブロックを搭載したデバイス・アーキテクチャの場合、ROM コードを
推測するためにアドレスまたは出力をラッチする必要があります。出力
レジスタが使用される場合、これらのレジスタは Stratix RAMブロック
の入力レジスタを使用して実装されますが、ROM の機能は変更されま
せん。アドレスをラッチする場合、推測される ROMのパワーアップ状
態が HDLデザインと異なる場合があります。このようなシナリオでは、
一般に合成ソフトウェアが警告を発行します。Quartus II ヘルプは、
Quartus II 合成機能を使用するときに機能が変更される状況を説明して
います。これらの ROM コード・サンプルは、アルテラ TriMatrix メモ
リ・アーキテクチャに直接マップしています。

例 6–19. Verilog HDL同期 ROM
module sync_rom (clock, address, data_out);

input clock;
input [7:0] address;
output [5:0] data_out;

reg [5:0] data_out;

always @ (posedge clock)
begin

case (address)
8'b00000000: data_out = 6'b101111;
8'b00000001: data_out = 6'b110110;
...
8'b11111110: data_out = 6'b000001;
8'b11111111: data_out = 6'b101010;

endcase
end

endmodule

例 6–20. VHDL同期 ROM
LIBRARY ieee;
USE ieee.std_logic_1164.ALL;

ENTITY sync_rom IS
PORT (

clock: IN STD_LOGIC;
address: IN STD_LOGIC_VECTOR(7 downto 0);
data_out: OUT STD_LOGIC_VECTOR(5 downto 0)

);
END sync_rom;

ARCHITECTURE rtl OF sync_rom IS
BEGIN
PROCESS (clock)

BEGIN
IF rising_edge (clock) THEN

CASE address IS
WHEN "00000000" => data_out <= "101111";
WHEN "00000001" => data_out <= "110110";
...
WHEN "11111110" => data_out <= "000001";
WHEN "11111111" => data_out <= "101010";
WHEN OTHERS => data_out <= "101111";

END CASE;
END IF;
END PROCESS;

END rtl;

altshift_taps—シフト・レジスタを HDLコードから推測

シフト・レジスタを推測する場合、合成ツールは同じ長さのシフト・レ
ジスタ・グループを検出して、これらを altshift_tapsメガファンク
ションに変換します。すべてのシフト・レジスタが検出されるには、以
下の特性を備えている必要があります。

■ 同じクロックおよびクロック・イネーブルを使用
■ 他のセカンダリ信号がない
■ タップがレジスタ 3個以上分離して等間隔に配置されている。

フォーマル検証ツールはシフト・レジスタ・メガファンクション
をサポートしないため、Quartus II 合成機能はサードパーティ・
フォーマル検証ツールが選択されている場合は、altshift_taps
メガファンクションを推測しません。Settingsダイアログ・ボッ
クスの EDA Tool Settingsページで、Quartus IIプロジェクトで
使用する EDAツールを選択できます。

フォーマル検証フローを使用している場合、アルテラではシフト・レジ
スタ・ロジックのみ搭載した別のエンティティまたはモジュールでシフ
ト・レジスタ・ブロックを作成することを推奨しています。これはフォー
マル検証中に、エンティティまたはモジュールをブラック・ボックスと
して処理する必要があるためです。

合成ソフトウェアは専用の RAM ブロックを搭載するデバイス・ファミ
リのシフト・レジスタのみ認識し、特定のガイドラインに従って最適な
実装を判断します。以下のガイドラインは Quartus II合成機能で使用さ
れ、一般にサードパーティ EDAツールでも使用されます。

■ FLEX® 10Kおよび ACEX® 1Kデバイスの専用メモリ量は比較的小さい
ため、これらのデバイスの場合、altshift_tapsメガファンクショ
ンは推測されません。

■ APEX 20K および APEX II デバイスの場合、ソフトウェアはシフト・
レジスタの総ビット数が 128 ビットを超える場合にのみ、
altshift_taps メガファンクションを推測します。シフト・レジ
スタが小さい場合、通常は専用メモリへの実装で効果は得られません。

■ Stratix および Cyclone™ シリーズのデバイスの場合、バスの幅（W）、
各タップ間の長さ（L）、およびタップ数（N）に基づいて
altshift_taps メガファンクションを推測するかどうかを判断し
ます。

● バス幅が 1 (W = 1)の場合、タップ数とタップ間の長 さを乗算し
た値が 64以上 (N × L ≥ 64)であれば、altshift_tapsが推
測されます。

● バス幅が 1を超える場合 (W > 1)、バス幅、タップ数、各タップ
間の長さを乗算した値が 32 以上であれば (W × N × L ≥ 32)、
altshift_tapsが推測されます。

タップ間の長さ（L）が 2 の累乗にならない場合、リード・カウンタと
ライト・カウンタをデコードする別のロジックが使用されます。このよ
うな状況になるのは、サイズの異なるシフト・レジスタの場合、ファン
クションの実装にロジック・エレメント（LE）またはアダプティブ・ロ
ジック・モジュール（ALM）を使用する外部デコード・ロジックが必要
になるためです。このデコード・ロジックによって、シフト・レジスタ
をメモリに実装する場合の性能上および利用上の利点が損なわれます。

altshift_taps メガファンクションにマップされ、RAM に配置され
るレジスタは、合成後にはそれらのノード名が存在しないため、シミュ
レーション・ツールの Verilog HDL または VHDL ファイルでは使用で
きません。

以下の例ではシフト・レジスタが推測されます。

■ Verilog HDLシングル・ビット幅、64ビット長シフト・レジスタ
■ タップが等間隔で配置された Verilog HDL 8ビット幅、64ビット長シ
フト・レジスタ

■ タップが等間隔で配置された VHDL 8ビット幅、64ビット長シフト・
レジスタ

Verilog HDLシングル・ビット幅、64ビット長シフト・レジスタ

例 6–21に示す Verilog HDLコード・サンプルは、シンプルなシングル・
ビット幅、64ビット長のシフト・レジスタを示しています。合成ソフト
ウェアは、サポートされるデバイスに対して、altshift_taps メガ
ファンクションにレジスタ（W = 1 と M  = 64）を実装します。レジス
タの長さが 64ビット未満の場合、シフト・レジスタはロジックに実装さ
れます。

例 6–21. Verilog HDLシングル・ビット幅、64ビット長シフト・レジスタ
module shift_1x64 (clk, shift, sr_in, sr_out);
  input clk, shift;

input sr_in;
output sr_out;

reg [63:0] sr;

always @ (posedge clk)
begin

if (shift == 1'b1)
begin

sr[63:1] <= sr[62:0];
sr[0] <= sr_in;

end
end
assign sr_out = sr[63];

endmodule

タップが等間隔で配置された Verilog HDL 8ビット幅、64ビット
長シフト・レジスタ

例 6–22と 6–23に示すコード・サンプルには、タップが 15、31、47番目
と等間隔で配置された Verilog HDL および VHDL 8 ビット幅、64 ビッ
ト長のシフト・レジスタ（W > 1とM = 64）を示しています。合成ソフ
トウェアはこのファンクションを 1つの altshift_tapsメガファンク
ションに実装し、それをサポートされるデバイスの RAM にマップしま
す。

例 6–22.タップが等間隔で配置された Verilog HDL 8ビット幅、64ビット長シフト・レジスタ
module shift_8x64_taps (clk, shift, sr_in, sr_out, sr_tap_one, sr_tap_two, sr_tap_three );

input clk, shift;
input [7:0] sr_in;
output [7:0] sr_tap_one, sr_tap_two, sr_tap_three, sr_out;

reg [7:0] sr [63:0];
integer n;

 always @ (posedge clk)
begin

if (shift == 1'b1)
begin

for (n = 63; n>0; n = n-1)
begin

sr[n] <= sr[n-1];
end 
sr[0] <= sr_in;

end 

end 
assign sr_tap_one = sr[15];
assign sr_tap_two = sr[31];
assign sr_tap_three = sr[47];
assign sr_out = sr[63];

endmodule

例 6–23.タップが等間隔で配置された VHDL 8ビット幅、64ビット長シフト・レジスタ
LIBRARY IEEE;
USE IEEE.STD_LOGIC_1164.ALL;
ENTITY shift_8x64_taps IS

PORT (
clk: IN STD_LOGIC;
shift: IN STD_LOGIC;
sr_in: IN STD_LOGIC_VECTOR(7 DOWNTO 0);
sr_tap_one: OUT STD_LOGIC_VECTOR(7 DOWNTO 0);
sr_tap_two : OUT STD_LOGIC_VECTOR(7 DOWNTO 0);
sr_tap_three: OUT STD_LOGIC_VECTOR(7 DOWNTO 0);
sr_out: OUT STD_LOGIC_VECTOR(7 DOWNTO 0)

);
END shift_8x64_taps;

ARCHITECTURE arch OF shift_8x64_taps IS
SUBTYPE sr_width IS STD_LOGIC_VECTOR(7 DOWNTO 0);
TYPE sr_length IS ARRAY (63 DOWNTO 0) OF sr_width;
SIGNAL sr: sr_length;

BEGIN
PROCESS (clk)
BEGIN

IF (clk'EVENT and clk = '1') THEN
IF (shift = '1') THEN

sr(63 DOWNTO 1) <= sr(62 DOWNTO 0);
sr(0) <= sr_in;

END IF;
END IF;

END PROCESS;
sr_tap_one <= sr(15);
sr_tap_two <= sr(31);
sr_tap_three <= sr(47);
sr_out <= sr(63);

END arch;

デバイス
固有の
コーディング・
ガイドライン

この項では、アルテラ・デバイスのアーキテクチャに対するデバイス固
有のコーディングの推奨事項について説明します。作成した HDL コー
ドを合成ツールがどのようにターゲットのアルテラ・デバイスにマップ
するかを理解することが重要です。レジスタおよび特定のロジック構造
を、該当するアルテラ・デバイスのアーキテクチャに合わせて設計する
ことによって、デザインの結果の品質を大幅に改善できます。

アルテラ・デバイスのレジスタ・パワーアップ値

デバイス・コアのレジスタは、すべてのアルテラ・デバイスで常に Low
（0）ロジック・レベルでパワーアップします。ただし、レジスタが High
（1）ロジック・レベルでパワーアップしたのと同様に動作するようロジッ
クを実装する方法があります。

レジスタ・アーキテクチャでプリセットをサポートしないデバイス上で、
プリセット信号を使用する場合、合成ツールはプリセット信号をクリア
信号に変換することがありますが、それには合成で NOT ゲート・プッ
シュバックと呼ばれる最適化を実行する必要があります。NOT-ゲート・
プッシュバックにより、レジスタの入力および出力にインバータが追加
されるため、リセット状態とパワーアップ状態は High になりますが、
デバイスは期待どおりに動作します。この場合、合成ツールはパワーアッ
プ状態を伝えるメッセージを発行することがあります。レジスタ自体は
Lowでパワーアップしますが、レジスタ出力が反転されるので、すべて
のディスティネーションで到着する信号は Highです。

このような影響により、特定のリセット値（0 以外）を指定すると、合
成ツールはレジスタで利用できる同期クリア（aclr）信号を使用して、
NOTゲート・プッシュバックでHighビットを実装する場合があります。
その場合、レジスタは指定されたリセット値にパワーアップしているよ
うに見えます。この動作が見られるのは、デザインが FLEX 10KE また
は ACEXデバイスをターゲットにしている場合です。

デバイスでロード信号が使用できる場合、合成ツールは 1または 0の非
同期ロードを使用して1または0値のリセットを実装できます。合成ツー
ルが非同期ロード信号を使用する場合、NOTゲート・プッシュバックは
実行しないため、レジスタは 0ロジック・レベルでパワーアップします。 

詳しくは、該当するデバイス・ファミリのハンドブック、またはアルテ
ラ・ウェブサイトwww.altera.co.jpで該当するハンドブックを参照して
ください。

設計者は、通常デザインに対して明示的なリセット信号を使用します。
この信号によって、必ずしもパワーアップ時でなくても、リセット後に
すべてのレジスタが適切な値に強制されます。非同期リセットによって
ボードが安全な状態で動作できるようにデザインを作成しておき、リ
セットがアクティブな状態でデザインを立ち上げることができます。こ
れがデバイスのパワーアップ状態に依存しない、適切な方法です。

レジスタの非同期コントロール・ポートをドライブする前に、デバイス・
アーキテクチャの外部ロジックまたは組み合わせロジックを同期させる
ことによって、デザインの安定性を高め、潜在的なグリッチの発生を防
止することができます。

適切な同期デザイン方法について詳しくは、「Quartus II ハンドブック
Volume 1」の「Design Recommendations for Altera Devices」の章を参
照してください。

デザインに特定のパワーアップ状態を強制する場合は、合成ツールで使
用可能な合成オプションを使用します。Quartus II 合成機能により、
Power-Up Level ロジック・オプションを適用できます。またソース・
コードで altera_attributeアサインメントを指定したオプションも
適用可能です。このオプションを使用すると、合成ツールは実際にはコ
ア・レジスタのパワーアップ状態を変更できないため、合成で強制的に
NOT ゲート・プッシュバックが実行されます。

Quartus II 合成機能の Power-Up Level アサインメントは、特定のレジ
スタまたはデザインのエンティティ、モジュール、またはサブデザイン
に適用できます。この場合、適用されたブロック内のすべてのレジスタ
に対するアサインメントとなります。レジスタはデフォルトでは 0でパ
ワーアップするため、このアサインメントを使用すると、NOT ゲート・
プッシュバックを使用して、すべてのレジスタが 1でパワーアップする
ように設定することができます。

NOTゲート・プッシュバックをグローバル・アサインメントとし
て使用すると、多数のインバータが必要になるため、結果の品質
が多少低下する場合があることに注意してください。状況によっ
ては、イネーブルまたはセカンダリ・コントロール・ロジックの
推測によって問題が生じます。またこのようなデザインを、ASIC
または HardCopy®デバイスに移行するのが困難な場合もありま
す。初期化を行っている場合は、機能シミュレーションでパワー
アップ動作をシミュレートできます。

Power-Up Level オプションと altera_attribute については、
「Quartus II ハンドブック Volume 1」の「Quartus II 合成機能」の章で
説明しています。 

VHDLでは、合成ツールによっては、レジスタの初期化デバイスに実装
することも可能です。例えば、Quartus II合成機能は、レジスタの VHDL
デフォルト値を Power-Up Level 設定に変換します。このように合成さ
れた動作は、機能シミュレーションでの VHDLコードのパワーアップ状
態に一致します。
6–32 Altera Corporation
2006年 5月



デバイス 固有の コーディング・ガイドライン
例えば、以下のコードでは qのレジスタが推測され、そのパワーアップ・
レベルを High（リセット値は 0）に設定します。

SIGNAL q : STD_LOGIC := '1'; -- q has a default value of '1'

PROCESS (clk, reset)
BEGIN
   IF (reset = '1') THEN
      q <= '0';
   ELSIF (rising_edge(clk)) THEN
      q <= d;
   END IF;
END PROCESS;

ほとんどの合成ツールと同様に、Quartus IIソフトウェアはVerilog
HDL 初期ブロックを合成しません。したがって、このツールは
初期ブロックにおける変数への値の代入を合成しません。

クリア &クロック・イネーブルなどのセカンダリ・レジス
タ・コントロール信号

FPGA デバイス・アーキテクチャは、フリップフロップとしても知られ
るレジスタをベースにしています。アルテラ FPGAのレジスタは複数の
セカンダリ・コントロール信号（クリア信号やイネーブル信号など）を
提供しており、この信号を使用すると、特別なロジック・セルを使用し
ないで、各レジスタ用のコントロール・ロジックを実装できます。セカ
ンダリ信号のサポートはデバイス・ファミリごとに異なるため、デバイ
ス・ファミリのデータシートを参照して、ターゲット・デバイスで使用
可能な信号を確認してください。

デバイスで信号を最も効率的に活用するには、HDLコードが可能な限り
デバイス・アーキテクチャに一致している必要があります。アーキテク
チャの性質のために、コントロール信号には一定の優先順位が設定され
ているため、可能な場合はその優先順位に従って HDL コードを作成す
る必要があります。

合成ツールは通常のロジックを使用して、すべてのコントロール信号を
エミュレートできるため、常に適切な機能の結果が得られます。ただし、
コントロール信号が使用される条件および優先順位の点でデザイン要件
に柔軟性がある場合は、デザインをターゲット・デバイスのアーキテク
チャに一致させ、最も効率的な結果を達成してください。デザインの信
号の優先順位がターゲット・アーキテクチャと一致しない場合は、コン
トロール信号を実装するための特別なロジックが必要になることがあり
ます。この特別なロジックでは、さらに別のデバイス・リソースが使用
され、コントロール信号の遅延が増加する可能性があります。

また、状況によっては、デバイス・アーキテクチャで専用コントロール・
ロジック以外のロジックを使用すると大きな影響を及ぼす可能性があり
ます。例えば、クロック・イネーブル信号は、デバイス・アーキテクチャ
の同期リセット信号やクリア信号よりも優先順位が高くなります。クリ
ア信号は同期信号ですが、クロック・イネーブルはロジック・アレイ・
ブロック（LAB）のクロック・ラインをオフにします。このため、デバ
イス・アーキテクチャでは、同期クリアが有効になるのはクロック・エッ
ジが起こった場合に限定されます。クロック・イネーブル信号よりも優
先順位の高い同期クリア信号を使用してレジスタをコーディングする場
合、レジスタのデータ入力を使用して、クロック・イネーブル機能をエ
ミュレートする必要があります。信号はレジスタのクロック・イネーブ
ル・ポートを使用しないため、Clock Enable Multicycle制約は適用でき
ません。この場合、これらのコントロール信号の優先順位は、デバイス
で使用できる信号の優先順位に従うのが最適であるのは明らかであり、
異なる優先順位を使用すると、クロック・イネーブル信号へのアサイン
メントにより予測とは異なる結果が生じます。

アルテラ・デバイスのセカンダリ・コントロール信号の優先順位
は、他のベンダのデバイスの順序とは異なります。優先順位に関
するデザイン要件が柔軟な場合、FPGAベンダ間でデザインを移
行するときに、セカンダリ・コントロール信号がデザインの性能
要件に適合することを確認し、ターゲット・デバイスのアーキテ
クチャと整合するようにして最良の結果を達成してください。

アルテラのすべてのデバイス・ファミリで信号の順序は同じですが、前
述したとおり、すべてのデバイス・ファミリがすべての信号を供給する
とは限りません。以下の優先順序を遵守してください。

1. Asynchronous Clear, aclr—highest priority
2. Preset, pre
3. Asynchronous Load, aload
4. Enable, ena
5. Synchronous Clear, sclr
6. Synchronous Load, sload
7. Data In, data—lowest priority

以下の例では、前述した aclr、aload、ena コントロール信号を使用
してレジスタを作成する Verilog HDL および VHDL コードを示してい
ます。

 dff_all.vにはセンシティビティ・リストに adataが含まれてい
ませんが、dff_all.vhd は含まれています。これは Verilog HDL
言語の制限です。—非同期ロード信号を記述する方法はありませ
ん（aload が High の間、adata が切り替わると q が切り替わ
る）。このような制限が課せられていても、すべての合成ツール
はこの構造から aload信号を推測する必要があります。合成ツー
ルがこのような推測を実行する場合、合成ツールから情報または
警告メッセージが発行されることがあります。

例 6–24. ena、aclr、aloadコントロール信号を使用した Verilog HDL D型フリップフロップ
（レジスタ）
module dff_control(clk, aclr, aload, ena, data, adata, q);

input clk, aclr, aload, ena, data, adata;
output q;

reg q;

always @ (posedge clk or posedge aclr or posedge aload)
begin

if (aclr)
q <= 1'b0;

else if (aload)
q <= adata;

else
if (ena)

q <= data;
end

endmodule

例 6–25. ena、aclr、aloadコントロール信号を使用した VHDL D型フリップフロップ
（レジスタ）
LIBRARY ieee;
USE ieee.std_logic_1164.all;

ENTITY dff_control IS
PORT (

      clk: IN STD_LOGIC;
      aclr: IN STD_LOGIC;
      aload: IN STD_LOGIC;
      adata: IN STD_LOGIC;
      ena: IN STD_LOGIC;

 data: IN STD_LOGIC;
      q: OUT STD_LOGIC

);
END dff_control;

ARCHITECTURE rtl OF dff_control IS
BEGIN

PROCESS (clk, aclr, aload, adata)
BEGIN

        IF (aclr = '1') THEN
          q <= '0';
        ELSIF (aload = '1') THEN
          q <= adata;
        ELSE

IF (clk = '1' AND clk'event) THEN
IF (ena ='1') THEN

                q <= data;
END IF;

END IF;
END IF;

END PROCESS;
END rtl;

プリセット信号は、より柔軟な aload信号に置き換えられるため、多く
のデバイス・ファミリには備わっていません。したがって、この例には
プリセット信号は含まれていません。

異なる sloadおよび sclr信号を備えたレジスタを数多く作成すると、
sclrおよび sload信号は LABワイド信号（LAB内で共通の信号）な
ので、Quartus IIフィッタでレジスタを LABにパッキングするのが困難
になることがあります。また LAB ワイドの sload 信号を使用すると、
デバイスが持っている高速フィードバック・パスを使用してレジスタを
パッキングできなくなります。これは一部のレジスタは同一 LAB内の他
のロジックでパッキングできないことを意味します。

したがって、合成ツールはルック・アップ・テーブル（LUT）にスペー
スがある場合は通常、sloadまたは sclr信号の使用を避けます。LUT
が使用できる場合は、LUTを使用してこれらの信号を実装したほうが常

に柔軟性が高くなります。sload 信号および sclr 信号の用途は通常、
演算チェイン（カウンタ）、または良好な LABパッキングを可能にする
共通信号を備えた十分なレジスタ数を持つワイド・マルチプレクサなど
の特定のファンクションに限定されています。デバイス・ファミリごと
にコントロール信号の数が異なるため、これらの信号の推測もデバイス
独自に行われます。例えば、Stratix II デバイスはセカンダリ・コント
ロール信号については Stratixデバイスよりも柔軟なため、合成ツールは
より多くの Stratix IIデバイス用の sload および sclr信号を推測する
可能性があります。

これらの追加コントロール信号を使用する場合は、デバイス・アーキテ
クチャに合致する優先順序で使用してください。最も効率的な結果を達
成するためには、先の例の aclrが aloadよりも優先順位が高いのと同
様に、sclr信号の優先順位が sload信号よりも高くなるように設計し
てください。デザインが上記の条件を満たさない場合、これらのレジス
タ専用信号は推測されないことを覚えておいてください。このような場
合、Quartus IIソフトウェアは汎用ロジック・リソースを使って HDLで
記述されたとおりの回路を合成します。

Verilog HDL では、sload および sclr を用いた以下のコードは（モ
ジュール宣言にコントロール信号を追加した後）、例 6–24に示す Verilog
HDLの例の q <= data文を置き換えることができます。

例 6–26. Verilog HDL sload & sclr
if (sclr)

q <= 1'b0;
else if (sload)

q <= sdata;
else

q <= data;

VHDLでは、sloadと sclrを用いた以下のコードは（エンティティ宣
言にコントロール信号を追加した後）、例 6–25 に示す VHDL の例の
q <= data文を置き換えることができます。

例 6–27. VHDL sload & sclr
IF (sclr = '1') THEN

q <= '0';
ELSIF (sload = '1') THEN

q <= sdata;
ELSE

q <= data;

Tri-State Signals

アルテラ・デバイスをターゲットにしている場合、トップレベルの双方
向ピンまたは出力ピンに接続されているときにのみトライ・ステート信
号を使用する必要があります。下位レベルの双方向ピンは避け、出力ピ
ンまたは双方向ピンをドライブしている場合を除き、Z ロジック値を使
用するのは避けてください。

合成ツールはマルチプレクサ・ロジックを使用して、内部トライ・ステー
ト信号を含むデザインをアルテラ・デバイスに正しく実装しますが、ア
ルテラではこのコーディング方法は推奨していません。

階層ブロック・ベースまたはインクリメンタル・デザイン・フ
ローでは、下位レベルの双方向ポートが、他のデザイン・ロジッ
クに接続されずに、階層を通して直接トップレベルの出力ピンに
接続されている場合を除き、階層境界に双方向ポートを含めるこ
とはできません。下位レベル・ブロックで境界トライ・ステート
を使用する場合は、合成ソフトウェアによってトライ・ステート
が階層を通してトップレベルに押し上げられ、アルテラ・デバイ
スの出力ピンのトライ・ステート・ドライバが利用されます。ト
ライ・ステートを押し上げるには階層間の最適化が必要になるた
め、下位レベルのトライ・ステートはブロック・ベースのデザイ
ン手法で制限されます。 

例 6–28および 6–29に示すコード例は、トライ・ステート双方向信号を
作成する Verilog HDLと VHDLコードを示しています。

例 6–28. Verilog HDLトライ・ステート信号
module tristate (myinput, myenable, mybidir);

input myinput, myenable;
inout mybidir;
assign mybidir = (myenable ? myinput : 1'bZ);

endmodule

例 6–29. VHDLトライ・ステート信号
LIBRARY ieee;
USE ieee.std_logic_1164.ALL;
USE ieee.std_logic_arith.ALL;

ENTITY tristate IS
PORT (

mybidir : INOUT STD_LOGIC;
myinput : IN STD_LOGIC;
myenable : IN STD_LOGIC
);

END tristate;

ARCHITECTURE rtl OF tristate IS
BEGIN

mybidir <= 'Z' WHEN (myenable = '0') ELSE myinput;
END rtl;

アダー・ツリー

ターゲットにしたアルテラ・デバイスのアーキテクチャに合わせてア
ダー・ツリーを適切に構築すると、性能と集積度が大幅に改善される場
合があります。大きなアダー・ツリーを使用するアプリケーションの例
として、有限インパルス応答（FIR）フィルタがあります。パイプライ
ン化されたバイナリまたはターナリ・アダー・ツリーを適切に使用する
と、結果の品質を大幅に改善できます。

この項では、アルテラの 4入力 LUTデバイス、および現在 Stratix IIデ
バイスでのみ提供されている 6 入力 LUT ロジック構造について、コー
ディング推奨事項が異なる理由を説明します。

ロジック・エレメントに 4入力 LUTを使用したアーキテクチャ

Stratixシリーズ、Cycloneシリーズ、APEXシリーズ、および FLEXシ
リーズのデバイスなどのアーキテクチャは、LEの標準組み合わせ構造と
して 4入力 LUTを使用しています。

デザインがパイプライン化可能な場合、4 入力ルックアップ・テーブル
を使用するデバイスにおいて 3 つの数値（A、B、C）を最も高速で加算
する方法は、A + Bを計算し、出力をラッチし、ラッチされた出力を C
に加算することです。A + Bの加算には 1レベルのロジック（1つの LE
で 1ビットが加算される）が必要なので、最大クロック速度で実行され
ます。これは必要なだけ拡張できます。

例 6–30に示すコード・サンプルでは、5つの数字（A、B、C、D、E）が
追加されています。4 入力ルックアップ・テーブルを使用するデバイス
で 5つの数字を追加する場合、合計 64の LE（ 16ビット数値の場合）に
4つのアダーと 3レベルのレジスタが必要です。

例 6–30. Verilog HDLパイプライン・バイナリ・ツリー
module binary_adder_tree (A, B, C, D, E, CLK, OUT);

parameter WIDTH = 16;
input [WIDTH-1:0] A, B, C, D, E;
input CLK;
output [WIDTH-1:0] OUT;

wire [WIDTH-1:0] sum1, sum2, sum3, sum4;
reg [WIDTH-1:0] sumreg1, sumreg2, sumreg3, sumreg4;
// Registers

always @ (posedge CLK)
begin

sumreg1 <= sum1;
sumreg2 <= sum2;
sumreg3 <= sum3;
sumreg4 <= sum4;

end

// 2-bit additions
assign sum1 = A + B;
assign sum2 = C + D;
assign sum3 = sumreg1 + sumreg2;
assign sum4 = sumreg3 + E;
assign OUT = sumreg4;

endmodule

アダプティブ・ロジック・エレメントに 6 入力 LUT を使用した
アーキテクチャ

Stratix II アーキテクチャは基本ロジック構造に 6 入力 LUT を使用する
ため、Stratix IIデバイスは上記の 4入力 LUTを用いた例とは異なるコー
ディング・スタイルのほうが有利です。特に Stratix IIデバイスの ALM
は、3 ビットを同時に加算できます。したがって、上記の例のツリーは
深度が 2レベルで、4つの add-by-twoの代わりに、2つの add-by-three
で実現できることになります。

Stratix IIデバイスの上記の例のコードはコンパイルに成功しますが、こ
のコードは非効率であり、6 入力アダプティブ・ルック・アップ・テー
ブル（ALUT）を活用していません。ツリーをターナリ・ツリーとして
再構築することによって、デザインはより効率化され、集積度の利用率
も大幅に改善されます。したがって、Stratix IIデバイスをターゲットに

する場合、Stratix II デバイス・アーキテクチャを活用するには、4 入力
LUTアーキテクチャ向けにデザインされたパイプライン・バイナリ・ア
ダー・ツリーは書き直す必要があります。

例 6–31は Stratix IIデバイスの 32個の ALUTのみで実現できます。

このタイプのコーディング方法を使用している場合は LAB の入
力信号ライン数の制約のために、Stratix II LAB内のALMを加算
回路のみでは完全に使い切ること（パックする）ができない場合
があります。ただし、多くのデザインでは、Quartus II フィッタ
は他のロジックを空いている LABにパックして、未使用ALMを
活用します。

例 6–31. Verilog HDLパイプライン・ターナリ・ツリー
module ternary_adder_tree (A, B, C, D, E, CLK, OUT);

parameter WIDTH = 16;
input [WIDTH-1:0] A, B, C, D, E;
input CLK;
output [WIDTH-1:0] OUT;

wire [WIDTH-1:0] sum1, sum2;
reg [WIDTH-1:0] sumreg1, sumreg2;
// Registers

always @ (posedge CLK)
begin

sumreg1 <= sum1;
sumreg2 <= sum2;

end

// 3-bit additions
assign sum1 = A + B + C;
assign sum2 = sumreg1 + D + E;
assign OUT = sumreg2;

endmodule

上記の例では、パイプライン化された加算器を示していますが、加算演
算を分割すると非パイプライン化加算器でも結果が向上します。デザイ
ンがパイプライン化されていない場合、ターナリ・ツリーはバイナリ・
ツリーよりもはるかに性能が高くなります。例えば、合成ツールによっ
ては、HDLコード sum = (A + B + C) + (D + E)の方が括弧を使
用しないコードに比べ最適な実装、すなわち 3入力アダー（A + B + C
に対応）とそれに続く 3入力アダー（sum1 + D + Eに対応）が生成さ
れる可能性が高くなります。括弧を追加しない場合、合成ツールはアー
キテクチャに最適ではない方法で加算を分割する可能性があります。

その他の
ロジック
構造の
コーディング・
ガイドライン

この項では、以下のロジック構造のコーディングのガイドラインを示し
ます。

■ Latch—アルテラは可能であればLatchを使用しないことを推奨して
いますが、この項では Latch が必要な場合に、正しく使用するため
のガイドラインも記載しています。

■ ステート・マシン — この項は、ステート・マシンを使用するときに
最良の結果を得るのに役立ちます。

■ マルチプレクサ — この項では一般的な問題に対処し、マルチプレク
サ・デザインに対して最適なリソース利用を実現するためのデザイ
ン・ガイドラインを示しています。

■ CRC（Cyclic Redundancy Check）チェック機能—この項では、CRC
機能をデザインする場合に良好な結果を得るためのガイドラインを
示します。

Latch

Latch は、新しい値が割り当てられるまで信号の値を保持する、小さな
組み合わせループです。

アルテラでは、可能な場合はデザインに Latchを使用しないこと
を推奨しています。

Latch およびすべての組み合わせループを使用したデザインに関連する
問題について詳しくは、「Quartus IIハンドブック Volume 1」の「Design
Recommendations for Altera Devices」の章を参照してください。

Latchを使用する意図がないときは、「意図しない Latchの生成」で詳述
するように、HDLコードから Latchを推測できます。Latchを推測させ
る意図がある場合は、6–44ページの「Latchの正しい推測」で詳述する
ように、正しく推測して適切なデバイス動作を保証する必要があります。

意図しない Latchの生成

組み合わせロジックを設計する場合、特定のコーディング・スタイルで
は意図しない Latch が生成される可能性があります。例えば、CASE 文
または IF 文で可能なすべての条件に対応していないときには、新しい
出力値が割り当てられない場合に出力を保持する Latchが必要になるこ
とがあります。合成ツールのメッセージで、推測される Latchへの参照
を確認してください。コードで意図せずに Latch が作成された場合は、
コードの変更を行って Latchを削除します。

Latch はフォーマル検証ツールでのサポートが制限されていまし
た。したがって、デザイン・フローでフォーマル検証を使用して
いる場合は、不完全な CASE 文などを通じて、意図しない Latch
が推測されないようにしてください。

Verilog HDL デザインで full_case属性を使用すると、指定されてい
ない case を don’t care 値（X）として扱うことができます。ただし、
full_case 属性を使用すると、シミュレーションのミスマッチが生じ
る可能性があります。この属性は合成専用属性であり、シミュレーショ
ン・ツールは指定されていない caseを Latchとして扱うためです。

合成ツールでの属性の使用について詳しくは、「Quartus II ハンドブック
Volume 1」の「合成」セクションの該当する章を参照してください。
「Quartus II合成機能」の章には、可能性のあるシミュレーションのミス
マッチを説明した例が記載されています。

IF文または CASE文の最後の ELSE句または WHEN OTHERS句を省略し
ても、Latch が生成される可能性があります。デフォルト条件の Don’t
care（X）アサインメントは、Latchの生成を防止するのに有効です。最
高のロジック最適化を達成するには、デフォルトの CASE または最終的
な ELSE値を、ロジック値ではなく don’t care（X）に割り当ててください。

例 6–32に示す VHDLサンプル・コードでは、意図しない Latch生成が
防止されます。最後の ELSE句を省略すると、このコードでは sel入力
の残りの組み合わせに対応する意図しない Latchが作成されます。この
コードで Stratixデバイスをターゲットにする場合、最後の ELSE条件を
省略すると、合成ソフトウェアは最大 6 つの LE を使用してしまいます
が、ELSE文を使用すれば 3つで済みます。また、ELSE文で Xの値を割
り当てる代わりに 1 を割り当てると、合成ソフトウェアは don’t care値
の指定により、より良い最適化を実行できます。LEが増加することがあ
ります。

例 6–32.意図しない Latch作成を防ぐ VHDLコード
LIBRARY ieee;
USE IEEE.std_logic_1164.all;

ENTITY nolatch IS
PORT (a,b,c: IN STD_LOGIC;

sel: IN STD_LOGIC_VECTOR (4 DOWNTO 0);
oput: OUT STD_LOGIC);

END nolatch;

ARCHITECTURE rtl OF nolatch IS
BEGIN

PROCESS (a,b,c,sel) BEGIN
IF sel = "00000" THEN

oput <= a;
ELSIF sel = "00001" THEN

oput <= b;
ELSIF sel = "00010" THEN

oput <= c;
ELSE  --- Prevents latch inference

oput <= ''X'; --/
END IF;

END PROCESS;
END rtl;

Latchの正しい推測

合成ツールは、一般に組み合わせループに関連した問題を含まない安全
な Latchを推測できます。

Quartus II 合成機能を使用する場合、ソフトウェアで推測される Latch
はコンパイル・レポートのUser-Specified and Inferred Latchesセクショ
ンで報告されます。このレポートには、その Latchが安全かどうか、タ
イミング・ハザードがないかどうかが示されます。

設計したデザインの Latch または組み合わせループが、User-Specified
and Inferred Latches レポートに記載されていない場合、ソフトウェア
によって安全な Latchとして推測されておらず、グリッチ・フリーと見
なされていないことを意味します。

Compilation Reportの Analysis & Synthesis Logic Cells Representing
Combinational Loopsテーブルにリストされるすべての組み合わせルー
プは、タイミング・ハザードが危険な状態にあります。これらのエント
リは、調査が必要なデザインの問題があることを示しています。ただし、
組み合わせループを含む正しいデザインを作成することは可能です。例
えば、組み合わせループを非センシティブにすることは可能です。これ
はハードウェア中に存在する経路について、設計者はその経路をアク
ティブにするデータが回路に発生することがないこと、あるいはデータ

入力には関係なく、その経路がセンシティブになるのを防止する相互排
他的な方法で周辺ロジックがセットアップされていることを把握してい
ます。

MAX® 7000AE および MAX 3000A などのマクロセル・ベースのデバイ
スの場合、Analysis & Synthesis User-Specified and Inferred Latches
テーブルにリストされているすべての data（D型）Latchと set-reset（S-
R）Latchが、グリッチなどのタイミング・ハザードなしで実装されてい
ます。グリッチの発生を防ぐ補正項が実装され、1 つのマクロセルによ
りフィードバック・ループが形成されます。

Stratix デバイス、Cyclone シリーズ、および MAX II デバイスなどの 4
入力LUTベースのデバイスの場合、User-Specified and Inferred Latches
テーブルのフィードバック・ループに 1つの LUTを持つすべての Latch
は、1 つの入力のみが変化するのであればタイミング・ハザードは生じ
ません。ある 1つの入力が変化しても同じ出力値になるのであれば、出
力にはグリッチが発生しないという LUT の性質があるためです。例え
ば、イネーブル入力が非アクティブのときに D 型入力がトグルしたり、
または優先順位の高いリセット入力がアクティブのときに set 入力がト
グルする場合です。LUTのこのハードウェア動作は、1つの LUTのみで
構成されたループには補正項が不要であることを意味します。
Quartus II ソフトウェアは、可能な場合は常にフィードバック・ループ
で 1 つの LUT を使用します。入力にフィードバックされる出力に加え
て、data、enable、set、reset入力を備えた Latchは、1つの 4入力 LUT
には実装できません。入力が多すぎるため、Quartus II ソフトウェアが
1つの LUTループで Latchを実装できない場合は、User-Specified and
Inferred Latches テーブルに Latch にタイミング・ハザードがあること
が示されます。

Stratix IIなどの 6入力 LUTベースのデバイスの場合、1つのアダプティ
ブ・ルック・アップ・テーブル（ALUT）で、組み合わせループ内にす
べての Latch 入力を実装できます。したがって、User-Specified and
Inferred Latchesテーブルのすべての Latchでは、1つの入力のみが変化
するのであればタイミング・ハザードはありません。

ハザードのない動作を確実にするために、同時に変更できるのは 1つの
コントロール入力のみです。Set と reset を同時にデアサートするなど、
2 つの入力を同時に変更したり、または data と enable を同時に変更す
ると、Latchは不正な動作を行います。

Quartus II 合成機能では、Verilog HDL の always ブロック、および
VHDLの process文から Latchが推測されますが、Verilog HDLでの
連続アサインメントやVHDLの同時信号アサインメントからは推測され

ません。これらのルールはレジスタの推測の場合と同じです。レジスタ
またはフリップフロップは、alwaysブロックおよび process文から推測
されます。以下の例では Latchが推測されます。

Verilog HDL set-reset Latchの例

例 6–33 に示す Verilog HDL コードのサンプルでは、Quartus II ソフト
ウェアで S-R Latchが正しく推測されます。

例 6–33. Verilog HDL set-reset Latch
module simple_latch (
   input SetTerm,
   input ResetTerm,
   output reg LatchOut
   );

   always @ (SetTerm or ResetTerm) begin
      if (SetTerm)
         LatchOut = 1'b1;
      else if (ResetTerm)
         LatchOut = 1'b0;
      end
endmodule

HDL Data型 Latchの例

例 6–34に示す VHDLコードのサンプルでは、Quartus IIソフトウェア
で D型 Latchが正しく推測されます。

例 6–34. HDL Data型 Latch
LIBRARY IEEE;
USE IEEE.std_logic_1164.all;

ENTITY simple_latch IS
   PORT (
      enable, data    : IN STD_LOGIC;
      q               : OUT STD_LOGIC
   );
END simple_latch;

ARCHITECTURE rtl OF simple_latch IS
BEGIN

latch : PROCESS (enable, data)
   BEGIN
      IF (enable = '1') THEN
         q <= data;
      END IF;
   END PROCESS latch;
END rtl;

以下の例は、Quartus II ソフトウェアで Latch が推測されない Verilog
HDLの連続アサインメントを示します。動作は Latchに類似しています
が、Latchとしては正しく機能しない場合があり、Latchとしてのタイミ
ング解析は行われません。

assign latch_out = en ? data : latch_out;

Quartus II合成機能では、lpm_latchメガファンクションのインスタン
ス化が可能な場合は安全な Latchを作成します。このメガファンクショ
ンを使用して、data、enable、set、resetの各入力を組み合わせて Latch
を作成します。安全な Latchを作成する場合、HDLコードから Latchを
推測する場合と同じ制限が適用されます。

アルテラのlpm_latchファンクションを別の合成ツールで推測すると、
この実装も Quartus IIソフトウェアは Latchとして認識します。サード
パーティ合成ツールで、lpm_latch メガファンクションを使用して
Latchを実装する場合、Quartus II合成機能では、HDLソース・コード
で作成された Latch がリストされるのと同様に、User-Specified and
Inferred Latchesテーブルに Latchがリストされます。lpm_latch実装
の生成に必要なコーディング・スタイルは、使用する合成ツールによっ
て異なります。サードパーティ合成ツールによっては、推測された
lpm_latchファンクション数をリストする場合があります。

Quartus II 合成機能によって Latch が User-Specified and Inferred
Latchesテーブルにリストされ、Analysis & Synthesisで安全な Latchと
して実装される場合、フィッタでのフィジカル・シンセシス・ネットリ
ストの最適化により、ハザードのない性能が維持されます。

LUTベースのファミリの場合、フィッタはAnalysis & Synthesisで Latch
イネーブルとして識別される信号を含むコントロール信号に対して、グ
ローバル配線を使用します。場合によっては、グローバル挿入遅延によ
り、タイミング性能が低下することがあります。必要であれば、
Quartus II の Global Signal ロジック・オプションをオフにして、手動
でグローバル信号の使用を防ぐことができます。グローバルLatchイネー
ブルは、コンパイル・レポートの Global & Other Fast Signalsテーブル
にリストされます。

ステート・マシン

合成ツールは、Verilog HDLと VHDLのステート・マシンを認識し、エ
ンコードできます。この項では、ステート・マシンを使用する場合に最
高の結果を得るためのガイドラインを示します。合成ツールがHDLコー
ドをステート・マシンとして認識できるようにすると、ツールによる状
態変数の再コーディングが可能になり、結果の品質が向上します。また、

ツールでステート・マシンのプロパティを使用できるようになり、デザ
インの他の部分が最適化されます。合成でステート・マシンが認識され
るとき、多くの場合はデザインの面積と性能を改善できます。

最高の結果を達成するために、合成ツールは FPGA デバイスには one-
hotエンコーディングを、CPLDデバイスにはminimal-bitエンコーディ
ングを使用することがよくあります。ただし、実装の選択はステート・
マシンやデバイスごとに異なります。ステート・マシンのエンコーディ
ング方法を制御する具体的な方法については、使用する合成ツールのマ
ニュアルを参照してください。

Quartus II 合成機能でのステート・マシンのエンコーディングについて
詳しくは、「Quartus IIハンドブック Volume 1」の「Quartus II Integrated
Synthesis」の章にある「State Machine Processing」の項を参照してくだ
さい。

ステート・マシンの適切な認識と推測を確実にし、結果の品質の向上さ
せるために、アルテラでは Verilog HDL および VHDL の両方に適用さ
れる以下のガイドラインに従うことを推奨しています。

■ 合成で不要な Latch が生成されないように、ステート・マシンから得
られた出力にデフォルト値を割り当ててください。

■ 出力値の割り当てを含むすべての演算ファンクションおよびデータ・
パスから、ステート・マシンのロジックを切り離します。

■ デザインに複数の状態で使用される演算が含まれる場合、ステート・
マシン外部での演算を定義し、ステート・マシンの出力ロジックで
この値が使用されるようにします。

■ シンプルな非同期または同期リセットを使用して、定義済みのパ
ワーアップ状態にします。ステート・マシンのデザインに、非同期
リセットと非同期ロードの両方が存在するなど、より込み入ったリ
セット・ロジックが含まれる場合、Quartus IIソフトウェアはステー
ト・マシンを推測しないで、通常のロジックを生成します。

デバイスに問題があるためステート・マシンが不正な状態に移行した場
合、ステート・マシンが次にリセットされるまで、デザインの正しい機
能が停止する可能性があります。デフォルトでは、合成ツールはこのよ
うな状況への対処を提供しません。システムに何らかの障害がある場合、
ステート・マシン以外の他のレジスタでも同様の問題が発生します。
default または when others 句を指定しても、デザインが意図的に
この状態に移行しない場合は、この動作には影響しません。デフォルト・
ステートにより生成されたロジックは、合成ツールにより削除されます。

合成ツールによっては、default caseが存在しない場合はそれを挿入し、
不正な状態を処理するデザインのロジックを保持する安全なステート・
マシンを実装するオプションを提供するものがあります。ステート・マ
6–48 Altera Corporation
2006年 5月



その他の ロジック 構造の コーディング・ガイドライン
シンが何らかの理由で不正な状態に移行した場合、次のクロック・エッ
ジでリセット状態に戻ります。当然ながら、このオプションではステー
ト・マシンのみが保護され、この方法ではデザインの他のすべてのレジ
スタは保護されません。 

ステート・マシンを実装するためのツール固有のオプションについて詳
しくは、ツール・ベンダのマニュアルか、「Quartus II ハンドブック
Volume 1」の「合成」セクションの該当する章を参照してください。

以下の「Verilog HDLステート・マシン」および「VHDLステート・マ
シン」の 2つの項では、言語別ガイドラインとコーディング例を説明し
ます。

Verilog HDLステート・マシン

Verilog HDL ステート・マシンを適切に認識および推測するために、以
下の Verilog HDLの補足ガイドラインに従ってください。これらのガイ
ドラインの一部は、Quartus II 合成機能のみ対象としています。具体的
なコーディングの推奨事項については、使用する合成ツールのマニュア
ルを参照してください。

■ SystemVerilog規格を使用する場合は、列挙型を使用してステート・
マシンを記述します（6–52 ページの「SystemVerilog ステート・マ
シンのコーディング例」を参照）。

■ ステート・マシンの状態を、Verilog-1995および-2001の parameter

データ型で表し、パラメータを使用してステート・アサインメント
を作成します、（以下の「Verilog HDLステート・マシンのコーディ
ング例」を参照）。この実装によりステート・マシンを読みやすくな
り、コーディング時にエラーが発生するリスクが低減されます。

アルテラでは、next_state <= 0など、状態変数に整数値を
直接使用しないこと推奨しています。ただし、整数を使用し
ても Quartus II ソフトウェアで推測が防止されることはあり
ません。

■ 状態遷移ロジックにおいて以下の例のように算術演算が使用された
場合、Quartus IIソフトウェアではステート・マシンは推測されませ
ん。

case (state)
0: begin

if (ena) next_state <= state + 2;
else next_state <= state + 1;

end
1: begin
...

endcase

■ 状態変数が出力となる場合、Quartus IIソフトウェアではステート・
マシンは推測されません。

Verilog HDLステート・マシンのコーディング例

以下のモジュール verilog_fsmは、標準的な Verilog HDLステート・
マシンの実装の例です。

このステート・マシンには 5つの状態があります。非同期リセットは変
数ステートを state_0に設定します。in1と in2の和は、state_1と
state_2 のステート・マシンの出力になります。差（in1 - in2）も
state_1 と state_2 で使用されます。一時変数 tmp_out_0 と
tmp_out_1は、in1と in2の和と差を保存します。これらの一時変数
をステート・マシンのさまざまな状態で使用すると、相互排他的状態間
でリソース共有が適切に行われます。

例 6–35. Verilog-2001ステート・マシン
module verilog_fsm (clk, reset, in_1, in_2, out);

input clk;
input reset;
input [3:0] in_1;
input [3:0] in_2;output [4:0] out;
parameter state_0 = 3'b000;
parameter state_1 = 3'b001;
parameter state_2 = 3'b010;
parameter state_3 = 3'b011;
parameter state_4 = 3'b100;

reg [4:0] tmp_out_0, tmp_out_1, tmp_out_2;
reg [2:0] state, next_state;

always @ (posedge clk or posedge reset)
begin

if (reset)
state <= state_0;

else
state <= next_state;

end
always @ (state or in_1 or in_2)
begin

tmp_out_0 = in_1 + in_2;
tmp_out_1 = in_1 - in_2;
case (state)

state_0: begin
tmp_out_2 <= in_1 + 5'b00001;
next_state <= state_1;

end
state_1: begin

if (in_1 < in_2) begin
next_state <= state_2;
tmp_out_2 <= tmp_out_0;

end
else begin

next_state <= state_3;
tmp_out_2 <= tmp_out_1;

end
end
state_2: begin

tmp_out_2 <= tmp_out_0 - 5'b00001;
next_state <= state_3;

end
state_3: begin

tmp_out_2 <= tmp_out_1 + 5'b00001;
next_state <= state_0;

end
state_4:begin

tmp_out_2 <= in_2 + 5'b00001;
next_state <= state_0;

end
default:begin

tmp_out_2 <= 5'b00000;
next_state <= state_0;

end
endcase

end
assign out = tmp_out_2;

endmodule

このステート・マシンの同等の実装は、以下のように、parameterデー
タ型の代わりに ‘defineを使用すると達成できます。

‘define state_0 3'b000
‘define state_1 3'b001
‘define state_2 3'b010
‘define state_3 3'b011
‘define state_4 3'b100

このケースでは、state アサインメントと next_state アサインメン
トには、以下の例に示すように state_xではなく ‘state_xが割り当
てられます。

next_state <= ‘state_3;

‘define構造がサポートされていますが、Parameterデータ・タ
イプを使用すると合成を通じて状態名が維持されるため、アルテ
ラではこのパラメータの使用を強く推奨しています。

SystemVerilogステート・マシンのコーディング例

例 6–36 に示すモジュール enum_fsm は、列挙型を使用する System
Verilogステート・マシンの実装例です。アルテラでは、このコーディン
グ・スタイルを使用して SystemVerilog のステート・マシンを記述する
ことを推奨しています。

Quartus II 合成機能では、ステート・マシンの状態を定義する列
挙型は、例 6–36 に示すように符号なし整数型でなければなりま
せん。列挙型を int unsignedとして指定しない場合、デフォル
トで符号付き int型が使用されます。このケースでは、Quartus II
合成機能はデザインを合成しますが、ステート・マシンを認識し
たり、推測することはありません。

例 6–36.列挙型を使用する SystemVerilogステート・マシン
module enum_fsm (input clk, reset, input int data[3:0], output int o);

   enum int unsigned { S0 = 0, S1 = 2, S2 = 4, S3 = 8 } state, next_state;

   always_comb begin : next_state_logic
  next_state = S0;
  case(state)

S0: next_state = S1;
S1: next_state = S2;
S2: next_state = S3;
S3: next_state = S3;

  endcase
   end

   always_comb begin
  case(state) 

 S0: o = data[3];
 S1: o = data[2];
 S2: o = data[1];
 S3: o = data[0];

  endcase
   end

   always_ff@(posedge clk or negedge reset) begin
  if(~reset) 

 state <= S0;
  else 

 state <= next_state;  
   end
endmodule

VHDLステート・マシン

VHDLステート・マシンが正しく認識および推測されるように、ステー
ト・マシンの状態は列挙型で表し、対応する型を使用してステート・ア
サインメントを実行します。この実装によりステート・マシンを読みや
すくなり、コーディング時にエラーが発生するリスクが低減されます。
状態が列挙型で表されていない場合、合成ソフトウェア（Quartus II 合
成機能など）はステート・マシンを認識しません。代わりに、ステート・
マシンは通常のロジック・ゲートおよびレジスタとして実装され、ステー
ト・マシンは Compilation Reportの Analysis & Synthesis セクション
にステート・マシンとしてリストされません。

VHDLステート・マシンのコーディング例

以下のエンティティvhd1_fsmは、VHDLの通常のステート・マシンの
実装例です。
Altera Corporation 6–53
2006年 5月



Quartus IIハンドブック Volume 1
このステート・マシンには 5つの状態があります。非同期リセットは変
数状態を state_0 に設定します。in1 と in2 の和は、state_1 と
state_2 のステート・マシンの出力になります。差（in1 - in2）も
state_1 と state_2 で使用されます。一時変数 tmp_out_0 と
tmp_out_1は、in1と in2の和と差を保存します。これらの一時変数
をステート・マシンのさまざまな状態で使用すると、相互排他的状態間
でリソース共有が適切に行われます。

例 6–37. VHDLステート・マシン
LIBRARY ieee;
USE ieee.std_logic_1164.all;
USE ieee.numeric_std.all;

ENTITY vhdl_fsm IS
PORT(

clk: IN STD_LOGIC;
reset: IN STD_LOGIC;
in1: IN UNSIGNED(4 downto 0);
in2: IN UNSIGNED(4 downto 0);
out_1: OUT UNSIGNED(4 downto 0)
);

END vhdl_fsm;

ARCHITECTURE rtl OF vhdl_fsm IS
TYPE Tstate IS (state_0, state_1, state_2, state_3, state_4);
SIGNAL state: Tstate;
SIGNAL next_state: Tstate;

BEGIN
PROCESS(clk, reset)
BEGIN

IF reset = '1' THEN
state <=state_0;

ELSIF rising_edge(clk) THEN
state <= next_state;

END IF;
END PROCESS;
PROCESS (state, in1, in2)

VARIABLE tmp_out_0: UNSIGNED (4 downto 0);
VARIABLE tmp_out_1: UNSIGNED (4 downto 0);

BEGIN
tmp_out_0 := in1 + in2;
tmp_out_1 := in1 - in2;
CASE state IS

WHEN state_0 =>
out_1 <= in1;
next_state <= state_1;

WHEN state_1 =>
IF (in1 < in2) then

next_state <= state_2;
out_1 <= tmp_out_0;

ELSE
next_state <= state_3;
out_1 <= tmp_out_1;

END IF;

WHEN state_2 =>
IF (in1 < "0100") then

out_1 <= tmp_out_0;
ELSE

out_1 <= tmp_out_1;
END IF;

next_state <= state_3;
WHEN state_3 =>

out_1 <= "11111";
next_state <= state_4;

WHEN state_4 =>
out_1 <= in2;
next_state <= state_0;

WHEN OTHERS =>
out_1 <= "00000";
next_state <= state_0;

END CASE;
END PROCESS;

END rtl;

マルチプレクサ

多くの FPGAデザインで、マルチプレクサはロジック利用の大きな部分
を占めます。マルチプレクサ・ロジックを最適化すると、アルテラ・デ
バイスに最も効率的に実装することができます。この項では一般的な問
題に対処し、マルチプレクサ・デザインに対して最適なリソース利用を
実現するためのデザイン・ガイドラインを示しています。また各種のマ
ルチプレクサ、およびアルテラの Stratix デバイスなど、多くの FPGA
アーキテクチャで使用されている4入力LUTへのマルチプレクサの実装
方法について説明します。

Stratix II デバイスでは 6入力 LUT が使用されますが、ここでは
特に説明しません。最適化の原理と手法の多くは類似しています
が、Stratix II の 6入力 LUT デバイスではデバイス利用が異なり
ます。例えば、Stratix IIデバイスは、LEの 4入力 LUTに実装さ
れるものより幅の広いマルチプレクサを 1つの ALMに実装でき
ます。

マルチプレクサのタイプ

この最初の項では、さまざまなタイプの HDL コードからマルチプレク
サを作成する方法を説明します。CASE文、IF文、およびステート・マ
シンはすべて、デザインでのマルチプレクサ・ロジックの共通のソース
になります。これらの HDL 構造から、バイナリ・マルチプレクサ、セ
レクタ・マルチプレクサ、プライオリティ・マルチプレクサなど、さま
ざまなタイプのマルチプレクサが作成されます。HDLコードからのマル

チプレクサの作成方法および合成時の実装方法を理解することが、マル
チプレクサ構造を最適化し最良の結果を得るための最初の一歩になりま
す。

バイナリ・マルチプレクサ

バイナリ・マルチプレクサは、バイナリ・エンコード選択ビットに基づ
いて入力を選択します。以下の「Verilog HDL バイナリ・エンコード
Case文」の例は、簡単な 4:1バイナリ・マルチプレクサを記述する Verilog
HDLコードを示しています。

例 6–38. Verilog HDLバイナリ・エンコード Case文
case (sel)

2'b00: z = a;
2'b01: z = b;
2'b10: z = c;
2'b11: z = d;

endcase

4:1 バイナリ・マルチプレクサは、2つの 4入力 LUTを使用すると効率
的に実装されます。4:1マルチプレクサを使用する大規模なバイナリ・マ
ルチプレクサを構築できます。4:1 マルチプレクサのツリーから N 入力
マルチプレクサ（N:1マルチプレクサ）を構築すると、わずか 0.66*（N - 1）
の LUTを使用した構造となります。

セレクタ・マルチプレクサ

セレクタ・マルチプレクサには、各データ入力に個別の選択ラインがあ
ります。マルチプレクサの選択ラインは、one-hotエンコードされます。
以下の「Verilog HDL one-hot エンコード Case 文」の例には、one-hot
セレクタ・マルチプレクサを記述する簡単な Verilog HDLコードの例を
示しています。

例 6–39. Verilog HDL one-hotエンコード Case文
case (sel)

4'b0001: z = a;
4'b0010: z = b;
4'b0100: z = c;
4'b1000: z = d;
default: z = 1'bx;

endcase

セレクタ・マルチプレクサは、AND ゲートと OR ゲートのツリーとして
一般に構築されます。この方式を使用すると、2つの ANDゲートと 1つ
の ORゲートを使用する 1つの 4入力 LUTの 2本の選択ラインを使用し

て、2つの入力を選択できます。これらの LUTの出力は、ワイド ORゲー
トと組み合わせることができます。この構造の N-入力セレクタ・マルチ
プレクサには、最低でも 0.66*（N-0.5）の LUTが必要です。これは最良
のバイナリ・マルチプレクサよりもわずかに多くなります。

プライオリティ・マルチプレクサ

プライオリティ・マルチプレクサでは、選択ロジックが優先順位を示し
ます。信号の優先順位に基づく特定の順序で、正しい項目を選択するオ
プションをチェックする必要があります。これらの構造は一般に、VHDL
または Verilog HDLの IF、ELSE、WHEN、SELECTまたは ?:文から作
成されます。「優先順位を示す VHDL IF文」の項の VHDLコード例は、
図 6-2に示すような回路図を実装します。

例 6–40.優先順位を示す VHDL IF文
IF cond1 THEN z <= a;
ELSIF cond2 THEN z <= b;
ELSIF cond3 THEN z <= c;
ELSE z <= d;
END IF;

図 6-2 に示すマルチプレクサは、チェインを形成し、各条件または選択
ビットを一度に 1つずつ評価します。

図 6-2. IF文のプライオリティ・マルチプレクサの実装

N-入力プライオリティ・マルチプレクサは、チェイン内の 2:1 マルチプ
レクサごとに 1つの LUTを使用するため、N-1の LUTを必要とします。
マルチプレクサのこのチェインにより一般に遅延が増加します。これは
ロジックを通過するクリティカル・パスがチェイン内のすべてのマルチ
プレクサを横断しているためです。

マルチプレクサのタイミング遅延を改善するために、優先順位が要求さ
れない場合はプライオリティ・マルチプレクサの使用を回避してくださ
い。デザインで選択順序が重要ではない場合は、CASE文を使用して、プ
ライオリティ・マルチプレクサの代わりに、バイナリまたはセレクタ・
マルチプレクサを実装します。優先順位を要求する多重化デザインで、
構造全体での遅延が重要な場合は、デザインを再コーディングしてロ
ジック・レベル数を減らし、特にクリティカル・パスに沿った遅延を最
小化することを検討してください。

デフォルトまたはその他の Caseアサインメント

CASE文で caseを完全に指定するには、DEFAULT（Verilog HDL）アサ
インメントまたは OTHERS（VHDL）アサインメントを含めます。この
アサインメントは、選択ラインの多くの組み合わせが未使用の one-hot
エンコーディング方式で特に重要になります。未使用の選択ラインの組
み合わせに case を指定し、合成ツールにこれらの case の合成方法の情
報を提供してください。これは Verilog HDL および VHDL 言語仕様で
要求されます。

デザインによっては、主に設計者がこれらの caseが起こらないと想定す
るという理由から、未使用 caseの結果を考慮する必要がない場合があり
ます。これらのタイプのデザインでは、DEFAULTアサインメントまたは
OTHERS アサインメントに任意の値を選択できます。ただし、選択する
アサインメント値がデザインの実装に必要なロジック利用率に大きく影
響する可能性があることに注意してください。これはアサインメントの
値ごとに合成ツールの処理方法が異なり、また速度および面積の最適化
方法が異なるためです。

一般に、最良の結果を得るには、無効な case を定義済みの caseの 1つ
と組み合わせるのではなく、無効な CASE 選択を単独の DEFAULT 文ま
たは OTHERS文で明示的に定義します。

無効な caseの値が重要でない場合は、X（don’t care）ロジック値を割り
当てることによって、無効な caseを明示的に指定します。このアサイン
メントにより、合成ツールで最良の面積の最適化を実行できます。

HDLデザインおよび合成ツールに異なる DEFAULTまたは OTHERSアサ
インメントを使って実験し、デザインのロジック利用に及ぼす影響をテ
ストできます。

暗黙のデフォルト

Verilog HDLおよび VHDL の IF文は、CASE型では容易に対応できな
い条件を指定する便利な方法といえます。ただし、IF 文を使用すると、
マルチプレクサ・ツリーが複雑になって、容易に合成ツールで最適化で
きなくなります。

特に、すべての IF 文に、指定されていない場合でも、暗黙の ELSE 条
件が指定されます。これらの暗黙のデフォルトにより、多重化デザイン
では一層複雑になる可能性があります。

「暗黙のデフォルトの VHDL IF文」の以下のコードの例では、4入力（a、
b、c、d）と 1出力（z）のマルチプレクサを示しています。

例 6–41.暗黙のデフォルトの VHDL IF文
IF cond1 THEN

IF cond2 THEN
z <= a;

END IF;
ELSIF cond3 THEN

IF cond4 THEN
z <= b;

ELSIF cond5 THEN
z <= c;

END IF;
ELSIF cond6 THEN

z <= d;
END IF;

このコードは 4:1 マルチプレクサを実装しているようですが、コード内
の 3 つの IF 文のそれぞれには、指定されていない暗黙の ELSE 条件が
含まれています。ELSE caseの出力値が指定されていないため、合成ツー
ルはこれらのcaseに対して同じ出力値を維持することが意図されている
と仮定します。

例 6–42に示すコード・サンプルには、例 6–41に示すコードと同じ機能
を持つコードが示されていますが、ELSE case を明示的に指定していま
す。

例 6–42.デフォルト条件を明示的に指定した VHDL IF文
IF cond1 THEN

IF cond2 THEN
z <= a;

ELSE
z <= z;

END IF;
ELSIF cond3 THEN

IF cond4 THEN
z <= b;

ELSIF cond5 THEN
z <= c;

ELSE 
z <= z;

END IF;
ELSIF cond6 THEN

z <= d;
ELSE

z <= z;
END IF;

図 6-3は例 6–42のコードを表す回路図です。4つの入力しかありません
が、マルチプレクサ・ロジックが基本の 4:1 マルチプレクサよりもかな
り複雑であることを示しています。

図 6-3. 暗黙のデフォルトでの IF文のマルチプレクサ実装

多重化ロジックを簡略化し、不要なデフォルトを削除できる方法がいく
つかあります。最適な方法は、ロジックが 4:1 CASE文の構造を使用する
ようにデザインを再コーディングすることです。あるいは、優先順位が
重要な場合、コードを再構築してデフォルトの caseを推測し、マルチプ
レクサをフラット化することができます。この例では、IF cond1 THEN
IF cond2の代わりに、同じ機能を実行する IF （cond1 AND cond2）を
使用してください。また、デフォルトが don’t care caseかどうかを確認
します。この例では、他に有効な caseが起こらなければ、最後の ELSIF
cond6文を ELSE文に変更することができます。

マルチプレクサ・ロジック内の不要なデフォルト条件をなくすことで、
デザインの実装を簡単にし、ロジック利用率を低減することができます。

縮退マルチプレクサ

縮退マルチプレクサは、可能性のある caseのすべてが固有のデータ入力
に使用されていないマルチプレクサです。不要な caseは、これらのマル
チプレクサのロジック利用の効率低下の原因になる傾向があります。
degenerateマルチプレクサが、完全なバイナリ・マルチプレクサで効率
的なロジック利用を行えるように、このマルチプレクサを再コーディン
グできます。

バイナリ・マルチプレクサの選択ライン数によって通常、希望の機能を
実装するのに必要なマルチプレクサのサイズが決まります。例えば、
図 6-4に示すマルチプレクサ構造には、16入力を備えたバイナリ・マル
チプレクサを実装可能な 4本の選択ラインがあります。しかし、このデ
ザインは 16 入力をすべて使用しておらず、このマルチプレクサは 16:1
縮退マルチプレクサになります。

例 6–43.縮退マルチプレクサを記述する VHDL CASE文
CASE sel[3:0] IS

WHEN "0101" => z <= a;
WHEN "0111" => z <= b;
WHEN "1010" => z <= c;
WHEN OTHERS => z <= d;

END CASE;

図 6-4. バイナリ縮退マルチプレクサ

図 6-4の例では、トップ・レベルの 1番目と 4番目のマルチプレクサを
容易になくすことができます。これは各マルチプレクサの 4入力がすべ
て同じ値であり、図 6-5 に示すように、他のマルチプレクサの入力数を
減らすことができるためです。

図 6-5. 縮退バイナリ・マルチプレクサの最適化バージョン

このバージョンのマルチプレクサを実装する場合、残りの 3:1 マルチプ
レクサのそれぞれに 2個ずつ、2:1マルチプレクサに 1個と最低でも 5個
の 4入力 LUTが必要になります。このデサインは 4入力からのみ出力を
選択します。4:1バイナリ・マルチプレクサでは、最適に実装できるのは
2個の LUTなので、この縮退マルチプレクサ・ツリーではロジックの効
率が低下します。

この構造のロジック利用率は、完全な 4:1 バイナリ・マルチプレクサを
実装できるように選択ラインを再コーディングすることによって改善で
きます。例 6–44のコード・サンプルは、オリジナルの選択ラインをバイ
ナリ・エンコーディングにより 1つの z_selに変換するレコーダ・デザ
インを示しています。

例 6–44.縮退バイナリ・マルチプレクサ用 VHDLレコーダ・デザイン
CASE sel[3:0] IS

WHEN "0101" => z_sel <= "00";
WHEN "0111" => z_sel <= "01";
WHEN "1010" => z_sel <= "10";
WHEN OTHERS => z_sel <= "11";

END CASE;

例 6–45のコード・サンプルに、完全なバイナリ・マルチプレクサの実装
方法を示します。

例 6–45. VHDL4:1バイナリ・マルチプレクサ・デザイン
CASE z_sel[1:0] IS

WHEN "00" => z <= a;
WHEN "01" => z <= b;
WHEN "10" => z <= c;
WHEN "11" => z <= d;

END CASE;

レコーダ・デザインの新しい z_selコントロール信号を使用して、4つ
の入力 a、b、c、dのいずれかを選択する 4:1バイナリ・マルチプレクサ
を制御します（図 6-6を参照）。選択ラインの複雑さは Recoderデザイン
で処理され、データ・マルチプレクシングは最も効率的な実装を可能に
する、シンプルなバイナリ選択ラインで実行されます。

図 6-6. Recoderによるバイナリ・マルチプレクサ

このレコーダのデザインは 2つの LUTに実装でき、また効率的な 4:1バ
イナリ・マルチプレクサが 2つの LUT を使用するため、LUTの合計は
4つになります。オリジナルの縮退マルチプレクサは 5つの LUTを要求
していました。このため、再コーディング・バージョンで使用するロジッ
クはオリジナルよりも 20%少なくなります。

マルチプレクサのロジック利用率は、選択ラインを完全なバイナリ case
に再コーディングすると改善されることがよくあります。エンコーディ
ングを実行するためのロジックが必要ですが、多くの場合は全体的なロ
ジック利用率が改善されます。

マルチプレクサのバス

多くの場合、マルチプレクサの入力は、一連のデータ入力バスで同じマ
ルチプレクサ機能が実行されるデータ入力バスです。このようなケース
では、マルチプレクサの非効率性はバスのビット数で乗算されます。上
記の項で説明した問題は、ワイド・マルチプレクサ・バスの場合はより
重要になります。

例えば、選択ラインを上記の項で詳述した完全なバイナリ caseに再コー
ディングする方法は、多重化バスでよく使用されます。選択ラインの再
コーディングは、バスのすべてのマルチプレクサに対して 1回のみ実行
する必要があります。バスのすべてのビットが Recoderロジックを共有
するため、マルチプレクサのバスのロジック効率を大幅に改善できます。

上記の項にある縮退マルチプレクサは、実装に 5つの LUTを必要としま
す。入力と出力が 32 ビット幅の場合、ファンクションはバス全体で 32
× 5または 160の LUTを必要とします。このレコーダ・デザインは 2つの
LUTを使用しており、全体のバスに対して選択ラインを 1回のみ再コー
ディングする必要があります。4:1バイナリ・マルチプレクサは、バスの
ビットあたり 2 つの LE を必要とします。再コーディングしたバージョ
ンの合計ロジック利用は、オリジナル・バージョンの 160 LUTと比べた
場合、バス全体で 2 + (2 × 32)または 66の LUTになります。ワイド・マ
ルチプレクサ・バスでは、ロジックの節約がより重要になります。

縮退マルチプレクサを最適化する手法の使用、不要な暗黙のデフォルト
の削除、最適な DEFAULTまたは OTHERS caseの選択は、マルチプレク
サのバスを最適化するときに重要な役割を果たす場合があります。

Quartus IIソフトウェアのマルチプレクサ再構築のためのオプショ
ン

Quartus II 合成機能には、合成中にマルチプレクサのバスを抽出し、最
適化する Restructure Multiplexers ロジック・オプションがあります。
特定の状況では、このオプションにより、デザインの HDL コードを変
更しないで、説明した再コーディング機能の一部が自動的に実行されま
す。Optimization 手法が Balanced（ほとんどのデバイス・ファミリの
デフォルト）または Area に設定されている場合、デフォルトによって
このオプションはオンになります。

詳しくは、「Quartus IIハンドブック Volume 1」の「Quartus II Integrated
Synthesis」の章の「Restructure Multiplexers 」項を参照してください。

CRC（Cyclic Redundancy Check）機能

CRC（Cyclic Redundancy Check）計算は、データの破損を検出するた
めに通信プロトコルおよびストレージ・デバイスで非常に多く使用され
ます。これらのファンクションは非常に効果的であり、破損したデータ
が 32ビット CRC検査に合格できる確率は非常に低いものです。

CRC機能は、通常はワイド XORゲートを使用してデータを比較します。
合成ツールがこれらの XORゲートをフラット化および分解して、FPGA
LUTにロジックを実装する方法は、デザインの面積および性能の結果に
大きな影響を及ぼす可能性があります。XORゲートには莫大な数の妥当
なファクタリングの組み合わせを作成するキャンセレーション・プロパ
ティがあるため、合成ツールがデフォルトで常に最良の結果を選択でき
るとは限りません。

Stratix IIデバイスの 6入力 ALUTは、これらのデザインでは 4入力 LUT
よりも非常に有利です。CRC処理を行うデザインは、Stratix IIデバイス
において非常に高速で動作させることが可能です。

アルテラ・デバイスでの CRCデザインの結果の品質を向上させるには、
以下のガイドラインが役立ちます。

速度最適化による性能向上

合成ツールは XORゲートをフラット化し、面積とロジックのレベルの深
度を最小化します。Quartus II 合成機能などの合成ツールは、このよう
なロジック構造に対してはデフォルトで面積の最適化をターゲットにし
ています。したがって、深度の低減に的を絞るには、合成の最適化手法
を速度に設定します。 

深度のフラット化によって、面積が大幅に増加する場合があります。

カスケード・ステージの代わりに別の CRCブロックを使用

一部の設計者は、CRCデザインを最適化し、8ビットの 4ステージなど
のカスケード・ステージを使用します。このようなデザインでは、デー
タ幅に応じて、また必要に応じて（8、24、または 32ビット後の計算な
ど）中間計算が使用されます。このデザインは FPGAデバイスでは最適
ではありません。CRC デザインで実行できる XOR キャンセレーション
は、このファンクションでは最終結果を求めるのに、すべての中間計算
が必要ないことを意味します。したがって、中間計算の使用を強制する

と、ファンクションの実装に必要な面積が増加し、またカスケード接続
のためにロジック深度も増加します。通常は、デザインで必要なデータ
幅ごとに完全に独立した CRC ブロックを作成し、次にそれらをまとめ
て多重化して、特定のタイミングで適切なモードを選択するのが得策で
す。

ブロックのマージを許可しないで独立した CRCブロックを使用

合成ツールは、XOR ロジックでのファクタリング・オプションのため
に、通常は 2 つの異なる CRC ブロックでリソースを共有し、重複を抽
出することによって CRC デザインの最適化を試みる場合がよくありま
す。前述したとおり、CRC ロジックでは大幅な面積削減が行えますが、
各 CRC 機能を個別に最適化したときに有効に機能します。共通のデー
タ信号でドライブされる、または同じディスティネーション信号を供給
する複数の CRC機能がある場合は、重複抽出をチェックします。 

結果の品質に問題があり、2つの CRC機能がロジックを共有している場
合は、以下の方法のいずれかを使用して、ブロックが個別に合成される
ようにしてください。

■ 各 CRC ブロックを独立したプロジェクトとして合成し、次に各 CRC
ブロックに対する個別の VQMまたは EDIFネットリスト・ファイル
を書き出します。

● Quartus II合成機能を使用してVQMファイルを作成するには、
Processingメニューで Startをクリックし、Start VQM Writer
をクリックします。

■ インクリメンタル・コンパイル・デザイン・フローでは、各 CRC ブ
ロックを独立したデザイン・パーティションとして定義します。 

● 詳しくは、Quartus IIハンドブック、Volume 1の階層ベースお
よびチーム・ベースのデザインのための Quartus IIインクリメ
ンタル・コンパイルの章を参照してください。

■ 合成オプションを使用して、CRCブロックの階層境界を維持します。
● AssignmentsメニューでAssignment Editorをクリックします。

Preserve Hierarchical Boundaryを Firmに設定します。

可能な場合はレイテンシを使用

デザインで CRC 機能を実装するために複数のサイクルを使用できる場
合、レジスタを追加しデザインをリ・タイミングすると、面積の縮小、
性能の改善、および消費電力の削減に役立つことがあります。合成ツー
ルにリ・タイミング機能（Quartus IIソフトウェアの Perform gate-level
register retiming オプションなど）がある場合、入力時に特別なレジス

タ・バンクを挿入し、リ・タイミング機能でレジスタを移動させて結果
を向上させることができます。また、1/2幅の CRCユニットを構築し、
各クロック・サイクルでデータの半分を切り替えることができます。

未使用時の CRCブロックを無効にして消費電力を削減

CRC デザインは、入力に変化があると常にロジックがトグルするため、
ダイナミック消費電力が大幅に増加します。消費電力を節約するには、
CRC が必要とされないすべてのクロック・サイクルで、クロック・イ
ネーブルを使用して CRC 機能を無効にしてください。一部のデザイン
は、他のロジックの実行中には、数クロック・サイクルの間 CRC 結果
を検査しません。このような短時間でも、CRC機能を無効にすることが
重要です。

デバイス同期ロード（sload）信号を使用した初期化

多くの CRC デザインでは、演算前にデータを 1 に初期化する必要があ
ります。ターゲット・デバイスが sload信号の使用をサポートしている
場合は、演算前にそれを使用して、デザイン内のすべてのレジスタを 1

に設定する必要があります。sload信号の使用を可能にするには、6–33
ページの「クリア & クロック・イネーブルなどのセカンダリ・レジス
タ・コントロール信号」の項で説明するコーディング・ガイドラインに
従ってください。タイミング・クロージャ・フロアプランまたはチップ・
エディタでレジスタの等価性を検査し、信号が予測どおりに使用されて
いることを確認します。 

sload 信号を使用して、レジスタの実装を強制する必要がある場合、
「Introduction to Low-Level Primitives Design User Guide」の説明に従っ
て、低レベル・デバイスのプリミティブを使用できます。

まとめ コーディング・スタイルとメガファンクションの実装は、デザインの性
能に非常に大きな影響を及ぼす可能性があるため、デザイン・プロセス
の当初から、コーディング・スタイルをデバイス・アーキテクチャに合
わせることが重要です。デザインの性能および面積利用率を改善するに
は、この章で説明したコーディングの推奨事項に従って、メモリや DSP
ブロックなどの最新デバイス機能、およびターゲットとなるアルテラ・
デバイスのロジック・アーキテクチャを活用してください。

その他の最適化の推奨事項については、「Quartus IIハンドブックVolume
2」の「Area & Timing Optimization」 の章を参照してください。

